# Theory of Designs in Isabelle/UTP

Simon Foster Yakoub Nemouchi Frank Zeyda

April 27, 2018

## Abstract

This document describes a mechanisation of the UTP theory of designs in Isabelle/UTP. Designs enrich UTP relations with explicit precondition/postcondition pairs, as present in formal notations like VDM, B, and the refinement calculus. If a program's precondition holds, then it is guaranteed to terminate and establish its postcondition, which is an approach known as total correctness. If the precondition does not hold, the behaviour is maximally nondeterministic, which represents unspecified behaviour. In this mechanisation, we create the theory of designs, including its alphabet, signature, and healthiness conditions. We then use these to prove the key algebraic laws of programming. This development can be used to support program verification based on total correctness.

## Contents

1	Des	ign Signature and Core Laws	2
	1.1	Definitions	2
	1.2	Lifting, Unrestriction, and Substitution	4
	1.3	Basic Design Laws	5
	1.4	Sequential Composition Laws	7
	1.5	Preconditions and Postconditions	10
	1.6	Distribution Laws	10
	1.7	Refinement Introduction	12
2	Des	ign Healthiness Conditions	13
	2.1	H1: No observation is allowed before initiation	13
	2.2	H2: A specification cannot require non-termination	16
	2.3	Designs as $H1$ - $H2$ predicates	19
	2.4	H3: The design assumption is a precondition	22
	2.5	Normal Designs as $H1$ - $H3$ predicates	24
	2.6	H4: Feasibility	26
	2.7	UTP theory of Designs	27
	2.8	UTP theories	27
	2.9	Galois Connection	28
	2.10	Fixed Points	29
3	Des	ign Proof Tactics	32

4	Imperative Programming in Designs 3		
	4.1 Assignment	32	
	4.2 Guarded Commands	34	
	4.3 Alternation	34	
	4.4 Iteration	38	
	4.5 Let and Local Variables	41	
	4.6 Deep Local Variables	41	
	5 Design Weakest Preconditions		
6	6 Refinement Calculus		
7	Theory of Invariants 7.1 Operation Invariants		
8	Meta Theory for UTP Designs	46	

## 1 Design Signature and Core Laws

theory utp-des-core imports UTP.utp begin

UTP designs [2, 4] are a subset of the alphabetised relations that use a boolean observational variable ok to record the start and termination of a program. For more information on designs please see Chapter 3 of the UTP book [4], or the more accessible designs tutorial [2].

## 1.1 Definitions

Two named theorem sets exist are created to group theorems that, respectively, provide prepostcondition definitions, and simplify operators to their normal design form.

named-theorems ndes and ndes-simp

```
alphabet des-vars =
  ok :: bool
declare des-vars.defs [lens-defs]
```

The two locale interpretations below are a technicality to improve automatic proof support via the predicate and relational tactics. This is to enable the (re-)interpretation of state spaces to remove any occurrences of lens types after the proof tactics *pred-simp* and *rel-simp*, or any of their derivatives have been applied. Eventually, it would be desirable to automate both interpretations as part of a custom outer command for defining alphabets.

```
interpretation des-vars: lens-interp \lambda r. (ok<sub>v</sub> r, more r) apply (unfold-locales) apply (rule injI) apply (clarsimp) done interpretation des-vars-rel:
```

lens-interp  $\lambda(r, r')$ .  $(ok_v r, ok_v r', more r, more r')$ 

```
apply (unfold-locales)
apply (rule injI)
apply (clarsimp)
done
lemma ok-ord [usubst]:
  \$ok \prec_v \$ok'
  by (simp add: var-name-ord-def)
type-synonym '\alpha des = '\alpha des-vars-scheme
type-synonym ('\alpha, '\beta) rel-des = ('\alpha des, '\beta des) urel
type-synonym '\alpha hrel-des = ('\alpha des) hrel
translations
  (type)'\alpha des \le (type)'\alpha des-vars-scheme
  (type)'\alpha des \le (type)'\alpha des-vars-ext
  (type) ('\alpha, '\beta) rel-des <= (type) ('\alpha des, '\beta des) urel
  (type)'\alpha hrel-des \le (type)'\alpha des hrel
notation des-vars-child-lens (\Sigma_D)
lemma ok-des-bij-lens: bij-lens (ok +_L \Sigma_D)
  by (unfold-locales, simp-all add: ok-def des-vars-child-lens-def lens-plus-def prod.case-eq-if)
Define the lens functor for designs
definition lmap-des-vars :: ('\alpha \Longrightarrow '\beta) \Rightarrow ('\alpha \ des-vars-scheme \Longrightarrow '\beta \ des-vars-scheme) <math>(lmap_D)
where [lens-defs]: lmap-des-vars = lmap[des-vars]
lemma lmap\text{-}des\text{-}vars: vwb\text{-}lens f \implies vwb\text{-}lens (lmap\text{-}des\text{-}vars f)
  by (unfold-locales, auto simp add: lens-defs)
lemma lmap-id: lmap_D 1_L = 1_L
  by (simp add: lens-defs fun-eq-iff)
lemma lmap\text{-}comp: lmap_D (f;_L g) = lmap_D f;_L lmap_D g
  by (simp add: lens-defs fun-eq-iff)
The following notations define liftings from non-design predicates into design predicates using
alphabet extensions.
abbreviation lift-desr (\lceil - \rceil_D)
where [P]_D \equiv P \oplus_p (\Sigma_D \times_L \Sigma_D)
abbreviation lift-pre-desr (\lceil - \rceil_{D <})
where \lceil p \rceil_{D <} \equiv \lceil \lceil p \rceil_{<} \rceil_{D}
abbreviation lift-post-desr (\lceil - \rceil_{D>})
where [p]_{D>} \equiv [[p]_>]_D
abbreviation drop\text{-}desr (|-|_D)
where |P|_D \equiv P \upharpoonright_e (\Sigma_D \times_L \Sigma_D)
abbreviation dcond :: ('\alpha, '\beta) \ rel\ des \Rightarrow '\alpha \ upred \Rightarrow ('\alpha, '\beta) \ rel\ des \Rightarrow ('\alpha, '\beta) \ rel\ des
  ((3- \triangleleft - \triangleright_D / -) [52,0,53] 52)
where P \triangleleft b \triangleright_D Q \equiv P \triangleleft \lceil b \rceil_{D \triangleleft} \triangleright Q
```

```
definition design::('\alpha, '\beta) \ rel\ des \Rightarrow ('\alpha, '\beta) \ rel\ des \Rightarrow ('\alpha, '\beta) \ rel\ des \ (infixl \vdash 60) where [upred\ defs]: P \vdash Q = (\$ok \land P \Rightarrow \$ok' \land Q)
```

An rdesign is a design that uses the Isabelle type system to prevent reference to ok in the assumption and commitment.

```
definition rdesign::('\alpha, '\beta) \ urel \Rightarrow ('\alpha, '\beta) \ urel \Rightarrow ('\alpha, '\beta) \ rel-des \ (infixl \vdash_r 60) \ where [upred-defs]: <math>(P \vdash_r Q) = \lceil P \rceil_D \vdash \lceil Q \rceil_D
```

An idesign is a normal design, i.e. where the assumption is a condition

```
definition ndesign::'\alpha \ cond \Rightarrow ('\alpha, '\beta) \ urel \Rightarrow ('\alpha, '\beta) \ rel-des \ (infixl \vdash_n 60) where [upred-defs]: (p \vdash_n Q) = (\lceil p \rceil_{<} \vdash_r Q)
```

```
definition skip-d :: '\alpha \ hrel-des \ (II_D) where [upred-defs]: II_D \equiv (true \vdash_r II)
```

```
definition bot-d :: ('\alpha, '\beta) \text{ rel-des } (\bot_D) where [upred-defs]: \bot_D = (false \vdash false)
```

```
definition pre\text{-}design :: ('\alpha, '\beta) \ rel\text{-}des \Rightarrow ('\alpha, '\beta) \ urel \ (pre_D) where [upred\text{-}defs]: pre_D(P) = |\neg P[true,false/\$ok,\$ok']|_D
```

```
definition post-design :: ('\alpha, '\beta) rel-des \Rightarrow ('\alpha, '\beta) urel (post_D) where [upred-defs]: post_D(P) = |P[true, true/\$ok, \$ok']|_D
```

#### syntax

```
-ok-f :: logic \Rightarrow logic (-f [1000] 1000)
-ok-t :: logic \Rightarrow logic (-t [1000] 1000)
-top-d :: logic (\top_D)
```

#### translations

```
P^f \rightleftharpoons CONST\ usubst\ (CONST\ subst-upd\ CONST\ id\ (CONST\ ovar\ CONST\ ok)\ false)\ P
P^t \rightleftharpoons CONST\ usubst\ (CONST\ subst-upd\ CONST\ id\ (CONST\ ovar\ CONST\ ok)\ true)\ P
T_D \Longrightarrow CONST\ not-upred\ (CONST\ utp-expr.var\ (CONST\ ivar\ CONST\ ok))
```

## 1.2 Lifting, Unrestriction, and Substitution

```
lemma drop-desr-inv [simp]: |\lceil P \rceil_D|_D = P
 by (simp add: prod-mwb-lens)
lemma lift-desr-inv:
  fixes P :: ('\alpha, '\beta) \text{ rel-des}
  assumes \$ok \ \sharp \ P \ \$ok' \ \sharp \ P
 shows \lceil \lfloor P \rfloor_D \rceil_D = P
 have bij-lens (\Sigma_D \times_L \Sigma_D +_L (in\text{-}var\ ok +_L \ out\text{-}var\ ok) :: (-, '\alpha\ des\text{-}vars\text{-}scheme \times '\beta\ des\text{-}vars\text{-}scheme)
lens)
    (is bij-lens (?P))
  proof -
    have ?P \approx_L (ok +_L \Sigma_D) \times_L (ok +_L \Sigma_D) (is ?P \approx_L ?Q)
      apply (simp add: in-var-def out-var-def prod-as-plus)
      apply (simp add: prod-as-plus[THEN sym])
    apply (meson lens-equiv-sym lens-equiv-trans lens-indep-prod lens-plus-comm lens-plus-prod-exchange
des-vars-indeps(1)
    done
    moreover have bij-lens ?Q
```

```
by (simp add: ok-des-bij-lens prod-bij-lens)
    ultimately show ?thesis
      by (metis bij-lens-equiv lens-equiv-sym)
  qed
  with assms show ?thesis
    apply (rule-tac aext-arestr[of - in-var ok +_L out-var ok])
    apply (simp add: prod-mwb-lens)
    apply (simp)
  apply (metis alpha-in-var lens-indep-prod lens-indep-sym des-vars-indeps(1) out-var-def prod-as-plus)
    using unrest-var-comp apply blast
 done
qed
lemma unrest-out-des-lift [unrest]: out \alpha \sharp p \Longrightarrow out \alpha \sharp \lceil p \rceil_D
 by (pred\text{-}simp)
lemma lift-dist-seq [simp]:
  [P ;; Q]_D = ([P]_D ;; [Q]_D)
 by (rel-auto)
lemma lift-des-skip-dr-unit [simp]:
  (\lceil P \rceil_D ;; \lceil II \rceil_D) = \lceil P \rceil_D
  (\lceil II \rceil_D ;; \lceil P \rceil_D) = \lceil P \rceil_D
 by (rel-auto)+
lemma lift-des-skip-dr-unit-unrest: \$ok' \sharp P \Longrightarrow (P ;; \lceil II \rceil_D) = P
 by (rel-auto)
lemma state-subst-design [usubst]:
  [\sigma \oplus_s \Sigma_D]_s \dagger (P \vdash_r Q) = ([\sigma]_s \dagger P) \vdash_r ([\sigma]_s \dagger Q)
 by (rel-auto)
lemma design-subst [usubst]:
  \llbracket \$ok \sharp \sigma; \$ok ' \sharp \sigma \rrbracket \Longrightarrow \sigma \dagger (P \vdash Q) = (\sigma \dagger P) \vdash (\sigma \dagger Q)
 by (simp add: design-def usubst)
lemma design-msubst [usubst]:
  (P(x) \vdash Q(x))[x \to v] = (P(x)[x \to v] \vdash Q(x)[x \to v])
 by (rel-auto)
lemma design-ok-false [usubst]: (P \vdash Q)[false/\$ok] = true
 by (simp add: design-def usubst)
lemma ok-pre: (\$ok \land \lceil pre_D(P) \rceil_D) = (\$ok \land (\neg P^f))
 by (pred-auto robust)
lemma ok\text{-}post: (\$ok \land \lceil post_D(P) \rceil_D) = (\$ok \land (P^t))
 by (pred-auto robust)
        Basic Design Laws
1.3
lemma design-export-ok: P \vdash Q = (P \vdash (\$ok \land Q))
 by (rel-auto)
lemma design-export-ok': P \vdash Q = (P \vdash (\$ok' \land Q))
```

```
by (rel-auto)
lemma design-export-pre: P \vdash (P \land Q) = P \vdash Q
 by (rel-auto)
lemma design-export-spec: P \vdash (P \Rightarrow Q) = P \vdash Q
 by (rel-auto)
lemma design-ok-pre-conj: (\$ok \land P) \vdash Q = P \vdash Q
 by (rel-auto)
lemma true-is-design: (false \vdash true) = true
 by (rel-auto)
lemma true-is-rdesign: (false \vdash_r true) = true
 by (rel-auto)
lemma bot-d-true: \perp_D = true
 by (rel-auto)
lemma bot-d-ndes-def [ndes-simp]: \perp_D = (false \vdash_n true)
 by (rel-auto)
lemma design-false-pre: (false \vdash P) = true
 by (rel-auto)
lemma rdesign-false-pre: (false \vdash_r P) = true
 by (rel-auto)
lemma ndesign-false-pre: (false \vdash_n P) = true
 by (rel-auto)
lemma ndesign\text{-}miracle: (true \vdash_n false) = \top_D
 by (rel-auto)
lemma top-d-ndes-def [ndes-simp]: \top_D = (true \vdash_n false)
 by (rel-auto)
lemma skip-d-alt-def: II_D = true \vdash II
 by (rel-auto)
lemma skip-d-ndes-def [ndes-simp]: II_D = true \vdash_n II
 by (rel-auto)
lemma design-subst-ok:
  (P[true/\$ok] \vdash Q[true/\$ok]) = (P \vdash Q)
 by (rel-auto)
lemma design-subst-ok-ok':
  (P[true/\$ok] \vdash Q[true,true/\$ok,\$ok']) = (P \vdash Q)
proof -
 have (P \vdash Q) = ((\$ok \land P) \vdash (\$ok \land \$ok' \land Q))
   by (pred-auto)
 also have ... = ((\$ok \land P\llbracket true/\$ok \rrbracket) \vdash (\$ok \land (\$ok \land Q\llbracket true/\$ok \land \rrbracket) \llbracket true/\$ok \rrbracket))
  by (metis conj-eq-out-var-subst conj-pos-var-subst upred-eq-true utp-pred-laws.inf-commute ok-vwb-lens)
```

```
also have ... = ((\$ok \land P[true/\$ok]) \vdash (\$ok \land \$ok' \land Q[true,true/\$ok,\$ok']))
    by (simp add: usubst)
  also have ... = (P[true/\$ok] \vdash Q[true,true/\$ok,\$ok'])
    by (pred-auto)
  finally show ?thesis ..
qed
lemma design-subst-ok':
  (P \vdash Q[true/\$ok']) = (P \vdash Q)
proof -
 have (P \vdash Q) = (P \vdash (\$ok' \land Q))
   by (pred-auto)
  also have ... = (P \vdash (\$ok' \land Q[true/\$ok']))
   by (metis conj-eq-out-var-subst upred-eq-true utp-pred-laws.inf-commute ok-vwb-lens)
  also have ... = (P \vdash Q[true/\$ok'])
    by (pred-auto)
  finally show ?thesis ..
qed
1.4
         Sequential Composition Laws
theorem design-skip-idem [simp]:
  (II_D ;; II_D) = II_D
 by (rel-auto)
{\bf theorem}\ design\hbox{-}composition\hbox{-}subst\hbox{:}
  assumes
    $ok ′ ♯ P1 $ok ♯ P2
 shows ((P1 \vdash Q1) ;; (P2 \vdash Q2)) =
         (((\neg ((\neg P1) ;; true)) \land \neg (Q1 \llbracket true / \$ok' \rrbracket ;; (\neg P2))) \vdash (Q1 \llbracket true / \$ok' \rrbracket ;; Q2 \llbracket true / \$ok \rrbracket)))
proof
  have ((P1 \vdash Q1) ;; (P2 \vdash Q2)) = (\exists ok_0 \cdot ((P1 \vdash Q1) [ (ok_0) / (sok_0) / (sok_0) ]; (P2 \vdash Q2) [ (ok_0) / (sok_0) / (sok_0) ])
    by (rule seqr-middle, simp)
 also have ...
        = (((P1 \vdash Q1)[false/\$ok']]; (P2 \vdash Q2)[false/\$ok])
            \lor ((P1 \vdash Q1)[true/\$ok'] ;; (P2 \vdash Q2)[true/\$ok]))
    by (simp add: true-alt-def false-alt-def, pred-auto)
 also from assms
  have ... = (((\$ok \land P1 \Rightarrow Q1 \llbracket true/\$ok' \rrbracket) ;; (P2 \Rightarrow \$ok' \land Q2 \llbracket true/\$ok \rrbracket)) \lor ((\neg (\$ok \land P1)) ;;
true))
    by (simp add: design-def usubst unrest, pred-auto)
 \textbf{also have} \ ... = ((\neg\$ok \ ;; true_h) \lor ((\neg P1) \ ;; true) \lor (Q1 \llbracket true / \$ok \ \H] \ ;; (\neg P2)) \lor (\$ok \ \H \land (Q1 \llbracket true / \$ok \ \H] \ \r]
;; Q2[true/\$ok]))
    by (rel-auto)
 \textbf{also have} \ ... = (((\neg ((\neg P1); true)) \land \neg (Q1 \llbracket true/\$ok \' \rrbracket ;; (\neg P2))) \vdash (Q1 \llbracket true/\$ok \' \rrbracket ;; Q2 \llbracket true/\$ok \rrbracket)))
    by (simp add: precond-right-unit design-def unrest, rel-auto)
 finally show ?thesis.
qed
theorem design-composition:
 assumes
    \$ok' \sharp P1 \$ok \sharp P2 \$ok' \sharp Q1 \$ok \sharp Q2
 shows ((P1 \vdash Q1) ;; (P2 \vdash Q2)) = (((\neg ((\neg P1) ;; true)) \land \neg (Q1 ;; (\neg P2))) \vdash (Q1 ;; Q2))
  using assms by (simp add: design-composition-subst usubst)
```

theorem design-composition-runrest:

```
assumes
   \$ok \sharp P1 \$ok \sharp P2 ok \sharp\sharp Q1 ok \sharp\sharp Q2
 shows ((P1 \vdash Q1) ;; (P2 \vdash Q2)) = (((\neg ((\neg P1) ;; true)) \land \neg (Q1^t ;; (\neg P2))) \vdash (Q1 ;; Q2))
proof -
  have (\$ok \land \$ok' \land (Q1^t ;; Q2[true/\$ok])) = (\$ok \land \$ok' \land (Q1 ;; Q2))
  proof -
   have (\$ok \land \$ok' \land (Q1 ;; Q2)) = ((\$ok \land Q1) ;; (Q2 \land \$ok'))
     by (metis (no-types, lifting) conj-comm seqr-post-var-out seqr-pre-var-out)
   also have ... = ((Q1 \land \$ok') ;; (\$ok \land Q2))
     by (simp\ add:\ assms(3)\ assms(4)\ runrest-ident-var)
   also have ... = (Q1^t ;; Q2[true/\$ok])
      by (metis ok-vwb-lens seqr-pre-transfer seqr-right-one-point true-alt-def uovar-convr upred-eq-true
utp-pred-laws.inf.left-idem utp-rel.unrest-ouvar vwb-lens-mwb)
   finally show ?thesis
     by (metis utp-pred-laws.inf.left-commute utp-pred-laws.inf-left-idem)
  qed
  moreover have (\neg (\neg P1 ;; true) \land \neg (Q1^t ;; (\neg P2))) \vdash (Q1^t ;; Q2[true/\$ok]) =
                (\neg (\neg P1 :: true) \land \neg (Q1^t :: (\neg P2))) \vdash (\$ok \land \$ok' \land (Q1^t :: Q2[true/\$ok]))
   by (metis design-export-ok design-export-ok')
  ultimately show ?thesis using assms
   by (simp add: design-composition-subst usubst, metis design-export-ok design-export-ok')
qed
theorem rdesign-composition:
 ((P1 \vdash_r Q1) :: (P2 \vdash_r Q2)) = (((\neg ((\neg P1) :: true)) \land \neg (Q1 :: (\neg P2))) \vdash_r (Q1 :: Q2))
 by (simp add: rdesign-def design-composition unrest alpha)
theorem design-composition-cond:
  assumes
    out\alpha \sharp p1 \$ok \sharp P2 \$ok' \sharp Q1 \$ok \sharp Q2
 shows ((p1 \vdash Q1) ;; (P2 \vdash Q2)) = ((p1 \land \neg (Q1 ;; (\neg P2))) \vdash (Q1 ;; Q2))
  \mathbf{using}\ \mathit{assms}
  by (simp add: design-composition unrest precond-right-unit)
{\bf theorem}\ rdesign\hbox{-}composition\hbox{-}cond:
  assumes out\alpha \sharp p1
  shows ((p1 \vdash_r Q1) :: (P2 \vdash_r Q2)) = ((p1 \land \neg (Q1 :: (\neg P2))) \vdash_r (Q1 :: Q2))
  using assms
 by (simp add: rdesign-def design-composition-cond unrest alpha)
theorem design-composition-wp:
  assumes
   ok \sharp p1 \ ok \sharp p2
   \$ok \sharp Q1 \$ok' \sharp Q1 \$ok \sharp Q2 \$ok' \sharp Q2
  shows ((\lceil p1 \rceil_{<} \vdash Q1) ;; (\lceil p2 \rceil_{<} \vdash Q2)) = ((\lceil p1 \land Q1 \ wp \ p2 \rceil_{<}) \vdash (Q1 \ ;; \ Q2))
  using assms by (rel-blast)
theorem rdesign-composition-wp:
  ((\lceil p1 \rceil_{<} \vdash_r Q1) ;; (\lceil p2 \rceil_{<} \vdash_r Q2)) = ((\lceil p1 \land Q1 \ wp \ p2 \rceil_{<}) \vdash_r (Q1 ;; Q2))
  by (rel-blast)
theorem ndesign-composition-wp [ndes-simp]:
  ((p1 \vdash_n Q1) ;; (p2 \vdash_n Q2)) = ((p1 \land Q1 wp p2) \vdash_n (Q1 ;; Q2))
  by (rel-blast)
```

```
theorem design-true-left-zero: (true : ; (P \vdash Q)) = true
proof -
  have (true ;; (P \vdash Q)) = (\exists ok_0 \cdot true \llbracket \ll ok_0 \gg /\$ok' \rrbracket ;; (P \vdash Q) \llbracket \ll ok_0 \gg /\$ok \rrbracket)
    by (subst\ seqr-middle[of\ ok],\ simp-all)
  \textbf{also have} \ \dots = ((\textit{true}\llbracket \textit{false}/\$\textit{ok}\,\check{}\,\rrbracket \ ;; \ (P \vdash Q)\llbracket \textit{false}/\$\textit{ok}\,\rrbracket) \ \lor \ (\textit{true}\llbracket \textit{true}/\$\textit{ok}\,\check{}\,\rrbracket \ ;; \ (P \vdash Q)\llbracket \textit{true}/\$\textit{ok}\,\rrbracket))
    by (simp add: disj-comm false-alt-def true-alt-def)
  also have ... = ((true \llbracket false / \$ok' \rrbracket ;; true_h) \lor (true ;; ((P \vdash Q) \llbracket true / \$ok \rrbracket)))
    by (subst-tac, rel-auto)
  also have \dots = true
    by (subst-tac, simp add: precond-right-unit unrest)
  finally show ?thesis.
qed
theorem design-left-unit-hom:
  fixes P Q :: '\alpha hrel-des
  shows (II_D ;; (P \vdash_r Q)) = (P \vdash_r Q)
proof -
  have (II_D ;; (P \vdash_r Q)) = ((true \vdash_r II) ;; (P \vdash_r Q))
    by (simp add: skip-d-def)
  also have ... = (true \land \neg (II ;; (\neg P))) \vdash_r (II ;; Q)
  proof -
    have out\alpha \sharp true
      by unrest-tac
    \mathbf{thus}~? the sis
      using rdesign-composition-cond by blast
  qed
  also have ... = (\neg (\neg P)) \vdash_r Q
    by simp
  finally show ?thesis by simp
qed
theorem rdesign-left-unit [simp]:
  II_D :: (P \vdash_r Q) = (P \vdash_r Q)
  by (rel-auto)
theorem design-right-semi-unit:
  (P \vdash_r Q) :: II_D = ((\neg (\neg P) :: true) \vdash_r Q)
  by (simp add: skip-d-def rdesign-composition)
theorem design-right-cond-unit [simp]:
  assumes out\alpha \sharp p
  shows (p \vdash_r Q) ;; II_D = (p \vdash_r Q)
  using assms
  by (simp add: skip-d-def rdesign-composition-cond)
theorem ndesign-left-unit [simp]:
  II_D ; ; (p \vdash_n Q) = (p \vdash_n Q)
  by (rel-auto)
theorem design-bot-left-zero: (\perp_D ;; (P \vdash Q)) = \perp_D
  by (rel-auto)
theorem design-top-left-zero: (\top_D :: (P \vdash Q)) = \top_D
  by (rel-auto)
```

## 1.5 Preconditions and Postconditions

```
theorem design-npre:
  (P \vdash Q)^f = (\neg \$ok \lor \neg P^f)
 by (rel-auto)
theorem design-pre:
  \neg (P \vdash Q)^f = (\$ok \land P^f)
 by (simp add: design-def, subst-tac)
    (metis (no-types, hide-lams) not-conj-deMorgans true-not-false(2) utp-pred-laws.compl-top-eq
           utp-pred-laws.sup.idem utp-pred-laws.sup-compl-top)
theorem design-post:
  (P \vdash Q)^t = ((\$ok \land P^t) \Rightarrow Q^t)
 by (rel-auto)
theorem rdesign-pre [simp]: pre_D(P \vdash_r Q) = P
 by (pred-auto)
theorem rdesign-post [simp]: post_D(P \vdash_r Q) = (P \Rightarrow Q)
 by (pred-auto)
theorem ndesign-pre [simp]: pre_D(p \vdash_n Q) = \lceil p \rceil_{<}
 by (pred-auto)
theorem ndesign\text{-}post\ [simp]:\ post_D(p \vdash_n Q) = (\lceil p \rceil_{<} \Rightarrow Q)
 by (pred-auto)
lemma design-pre-choice [simp]:
  pre_D(P \sqcap Q) = (pre_D(P) \land pre_D(Q))
 by (rel-auto)
lemma design-post-choice [simp]:
  post_D(P \sqcap Q) = (post_D(P) \lor post_D(Q))
  by (rel-auto)
lemma design-pre-condr [simp]:
 pre_D(P \triangleleft \lceil b \rceil_D \triangleright Q) = (pre_D(P) \triangleleft b \triangleright pre_D(Q))
 by (rel-auto)
lemma design-post-condr [simp]:
  post_D(P \triangleleft \lceil b \rceil_D \triangleright Q) = (post_D(P) \triangleleft b \triangleright post_D(Q))
 by (rel-auto)
lemma preD-USUP-mem: pre_D (| | i \in A \cdot P i ) = ( | i \in A \cdot pre_D(P i) )
 by (rel-auto)
lemma preD-USUP-ind: pre_D (| | i \cdot P i) = ( | i \cdot pre_D(P i) )
 by (rel-auto)
        Distribution Laws
1.6
```

theorem design-choice:

$$(P_1 \vdash P_2) \sqcap (Q_1 \vdash Q_2) = ((P_1 \land Q_1) \vdash (P_2 \lor Q_2))$$
  
**by**  $(rel-auto)$ 

```
theorem rdesign-choice:
```

$$(P_1 \vdash_r P_2) \sqcap (Q_1 \vdash_r Q_2) = ((P_1 \land Q_1) \vdash_r (P_2 \lor Q_2))$$
 by  $(rel-auto)$ 

 $\textbf{theorem} \ ndesign\text{-}choice \ [ndes\text{-}simp]:$ 

$$(p_1 \vdash_n P_2) \sqcap (q_1 \vdash_n Q_2) = ((p_1 \land q_1) \vdash_n (P_2 \lor Q_2))$$
 by  $(\textit{rel-auto})$ 

**theorem** *ndesign-choice'* [*ndes-simp*]:

$$((p_1 \vdash_n P_2) \lor (q_1 \vdash_n Q_2)) = ((p_1 \land q_1) \vdash_n (P_2 \lor Q_2))$$
 by  $(rel-auto)$ 

**theorem** design-inf:

$$(P_1 \vdash P_2) \sqcup (Q_1 \vdash Q_2) = ((P_1 \lor Q_1) \vdash ((P_1 \Rightarrow P_2) \land (Q_1 \Rightarrow Q_2)))$$
 by  $(\textit{rel-auto})$ 

theorem rdesign-inf:

$$(P_1 \vdash_r P_2) \sqcup (Q_1 \vdash_r Q_2) = ((P_1 \lor Q_1) \vdash_r ((P_1 \Rightarrow P_2) \land (Q_1 \Rightarrow Q_2)))$$
 by  $(rel-auto)$ 

**theorem** ndesign-inf [ndes-simp]:

$$(p_1 \vdash_n P_2) \sqcup (q_1 \vdash_n Q_2) = ((p_1 \lor q_1) \vdash_n ((\lceil p_1 \rceil_{<} \Rightarrow P_2) \land (\lceil q_1 \rceil_{<} \Rightarrow Q_2)))$$
 **by**  $(rel-auto)$ 

theorem design-condr:

$$((P_1 \vdash P_2) \triangleleft b \triangleright (Q_1 \vdash Q_2)) = ((P_1 \triangleleft b \triangleright Q_1) \vdash (P_2 \triangleleft b \triangleright Q_2))$$
 by  $(rel-auto)$ 

**theorem** ndesign-dcond [ndes-simp]:

$$((p_1 \vdash_n P_2) \triangleleft b \triangleright_D (q_1 \vdash_n Q_2)) = ((p_1 \triangleleft b \triangleright q_1) \vdash_n (P_2 \triangleleft b \triangleright_r Q_2))$$
 by  $(rel-auto)$ 

 ${f lemma} \ design-UINF-mem:$ 

assumes 
$$A \neq \{\}$$

shows 
$$(\bigcap i \in A \cdot P(i) \vdash Q(i)) = (\coprod i \in A \cdot P(i)) \vdash (\bigcap i \in A \cdot Q(i))$$
 using assms by  $(rel-auto)$ 

**lemma** ndesign-UINF-mem [ndes-simp]:

assumes 
$$A \neq \{\}$$

shows 
$$(\bigcap i \in A \cdot p(i) \vdash_n Q(i)) = (\coprod i \in A \cdot p(i)) \vdash_n (\bigcap i \in A \cdot Q(i))$$
 using assms by  $(rel-auto)$ 

**lemma** *ndesign-UINF-ind* [*ndes-simp*]:

$$(\prod i \cdot p(i) \vdash_n Q(i)) = (\coprod i \cdot p(i)) \vdash_n (\prod i \cdot Q(i))$$
**by**  $(rel-auto)$ 

 $\mathbf{lemma}\ design\text{-}USUP\text{-}mem$ :

$$(\bigsqcup i \in A \cdot P(i) \vdash Q(i)) = (\prod i \in A \cdot P(i)) \vdash (\bigsqcup i \in A \cdot P(i) \Rightarrow Q(i))$$
 by  $(rel-auto)$ 

**lemma** ndesign-USUP-mem [ndes-simp]:

$$(\bigsqcup i \in A \cdot p(i) \vdash_n Q(i)) = (\prod i \in A \cdot p(i)) \vdash_n (\bigsqcup i \in A \cdot \lceil p(i) \rceil_{<} \Rightarrow Q(i))$$
 by  $(rel-auto)$ 

**lemma** ndesign-USUP-ind [ndes-simp]:

```
(\bigsqcup \ i \cdot p(i) \vdash_n Q(i)) = (\prod \ i \cdot p(i)) \vdash_n (\bigsqcup \ i \cdot \lceil p(i) \rceil_{<} \Rightarrow Q(i)) by (rel\text{-}auto)
```

## 1.7 Refinement Introduction

```
lemma ndesign-eq-intro:
  assumes p_1 = q_1 P_2 = Q_2
  shows p_1 \vdash_n P_2 = q_1 \vdash_n Q_2
  by (simp add: assms)
theorem design-refinement:
  assumes
    \$ok \ \sharp \ P1 \ \$ok' \ \sharp \ P1 \ \$ok \ \sharp \ P2 \ \$ok' \ \sharp \ P2
    \$ok \ddagger Q1 \$ok ' \ddagger Q1 \$ok \ddagger Q2 \$ok ' \ddagger Q2
  \mathbf{shows}\ (P1 \vdash Q1 \sqsubseteq P2 \vdash Q2) \longleftrightarrow (`P1 \Rightarrow P2` \land `P1 \land Q2 \Rightarrow Q1`)
proof -
  have (P1 \vdash Q1) \sqsubseteq (P2 \vdash Q2) \longleftrightarrow `(\$ok \land P2 \Rightarrow \$ok' \land Q2) \Rightarrow (\$ok \land P1 \Rightarrow \$ok' \land Q1)`
    by (pred-auto)
  also with assms have ... = (P2 \Rightarrow \$ok' \land Q2) \Rightarrow (P1 \Rightarrow \$ok' \land Q1)
    by (subst-subst-bool-split[of in-var ok], simp-all, subst-tac)
  also with assms have ... = (\neg P2 \Rightarrow \neg P1) \land ((P2 \Rightarrow Q2) \Rightarrow P1 \Rightarrow Q1)
    by (subst subst-bool-split[of out-var ok], simp-all, subst-tac)
  also have ... \longleftrightarrow '(P1 \Rightarrow P2)' \land 'P1 \land Q2 \Rightarrow Q1'
    by (pred-auto)
  finally show ?thesis.
qed
theorem rdesign-refinement:
  (P1 \vdash_r Q1 \sqsubseteq P2 \vdash_r Q2) \longleftrightarrow ('P1 \Rightarrow P2' \land 'P1 \land Q2 \Rightarrow Q1')
  by (rel-auto)
lemma design-refine-intro:
  assumes 'P1 \Rightarrow P2' 'P1 \land Q2 \Rightarrow Q1'
  shows P1 \vdash Q1 \sqsubseteq P2 \vdash Q2
  using assms unfolding upred-defs
  by (pred-auto)
lemma design-refine-intro':
  assumes P_2 \sqsubseteq P_1 \ Q_1 \sqsubseteq (P_1 \land Q_2)
  shows P_1 \vdash Q_1 \sqsubseteq P_2 \vdash Q_2
  using assms design-refine-intro [of P_1 P_2 Q_2 Q_1] by (simp add: refBy-order)
lemma rdesign-refine-intro:
  assumes 'P1 \Rightarrow P2' 'P1 \land Q2 \Rightarrow Q1'
  shows P1 \vdash_r Q1 \sqsubseteq P2 \vdash_r Q2
  using assms unfolding upred-defs
  by (pred-auto)
lemma rdesign-refine-intro':
  assumes P2 \sqsubseteq P1 \ Q1 \sqsubseteq (P1 \land Q2)
  shows P1 \vdash_r Q1 \sqsubseteq P2 \vdash_r Q2
  using assms unfolding upred-defs
  by (pred-auto)
lemma ndesign-refine-intro:
  assumes 'p1 \Rightarrow p2' '\lceil p1 \rceil_{<} \land Q2 \Rightarrow Q1'
```

```
shows p1 \vdash_n Q1 \sqsubseteq p2 \vdash_n Q2
 using assms unfolding upred-defs
 by (pred-auto)
lemma design-top:
 (P \vdash Q) \sqsubseteq \top_D
 by (rel-auto)
{f lemma}\ design\mbox{-}bottom:
 \perp_D \sqsubseteq (P \vdash Q)
 by (rel-auto)
\mathbf{lemma}\ design\text{-}refine\text{-}thms:
 assumes P \sqsubseteq Q
 shows 'pre_D(P) \Rightarrow pre_D(Q)' 'pre_D(P) \land post_D(Q) \Rightarrow post_D(P)'
 apply (metis assms design-pre-choice disj-comm disj-upred-def order-reft rdesign-refinement utp-pred-laws.le-iff-sup)
apply (metis assms conj-comm design-post-choice disj-upred-def refBy-order semilattice-sup-class le-iff-sup
utp-pred-laws.inf.coboundedI1)
done
end
\mathbf{2}
     Design Healthiness Conditions
theory utp-des-healths
 imports \ utp-des-core
begin
       H1: No observation is allowed before initiation
2.1
```

```
definition H1 :: ('\alpha, '\beta) \text{ rel-des} \Rightarrow ('\alpha, '\beta) \text{ rel-des} where
[upred-defs]: H1(P) = (\$ok \Rightarrow P)
lemma H1-idem:
  H1 (H1 P) = H1(P)
 by (pred-auto)
lemma H1-monotone:
  P \sqsubseteq Q \Longrightarrow H1(P) \sqsubseteq H1(Q)
 by (pred-auto)
lemma H1-Continuous: Continuous H1
 by (rel-auto)
lemma H1-below-top:
  H1(P) \sqsubseteq \top_D
 by (pred-auto)
lemma H1-design-skip:
  H1(II) = II_D
 by (rel-auto)
lemma H1-cond: H1(P \triangleleft b \triangleright Q) = H1(P) \triangleleft b \triangleright H1(Q)
 by (rel-auto)
```

```
lemma H1-conj: H1(P \land Q) = (H1(P) \land H1(Q))
 by (rel-auto)
lemma H1-disj: H1(P \lor Q) = (H1(P) \lor H1(Q))
 by (rel-auto)
lemma design-export-H1: (P \vdash Q) = (P \vdash H1(Q))
 by (rel-auto)
The H1 algebraic laws are valid only when \alpha(R) is homogeneous. This should maybe be gener-
alised.
{\bf theorem}\ \textit{H1-algebraic-intro}:
 assumes
   (true_h ;; R) = true_h
   (II_D ;; R) = R
 shows R is H1
proof -
 have R = (II_D ;; R) by (simp \ add: assms(2))
 also have \dots = (H1(II);; R)
   by (simp add: H1-design-skip)
 also have ... = ((\$ok \Rightarrow II) ;; R)
   by (simp add: H1-def)
 also have ... = (((\neg \$ok) ;; R) \lor R)
   by (simp add: impl-alt-def seqr-or-distl)
 also have ... = ((((\neg \$ok) ;; true_h) ;; R) \lor R)
   by (simp add: precond-right-unit unrest)
 also have ... = (((\neg \$ok) ;; true_h) \lor R)
   by (metis\ assms(1)\ seqr-assoc)
 also have ... = (\$ok \Rightarrow R)
   by (simp add: impl-alt-def precond-right-unit unrest)
 finally show ?thesis by (metis H1-def Healthy-def')
qed
lemma nok-not-false:
 (\neg \$ok) \neq false
 by (pred-auto)
theorem H1-left-zero:
 assumes P is H1
 shows (true ;; P) = true
proof -
 from assms have (true ;; P) = (true ;; (\$ok \Rightarrow P))
   by (simp add: H1-def Healthy-def')
 also from assms have ... = (true ;; (\neg \$ok \lor P)) (is - = (?true ;; -))
   by (simp add: impl-alt-def)
 also from assms have ... = ((?true ;; (\neg \$ok)) \lor (?true ;; P))
   using seqr-or-distr by blast
 also from assms have ... = (true \lor (true ;; P))
   by (simp add: nok-not-false precond-left-zero unrest)
 finally show ?thesis
   by (simp add: upred-defs urel-defs)
```

theorem H1-left-unit:

```
fixes P :: '\alpha \ hrel-des
 assumes P is H1
 shows (II_D ;; P) = P
proof -
 have (II_D ;; P) = ((\$ok \Rightarrow II) ;; P)
   by (metis H1-def H1-design-skip)
 also have ... = (((\neg \$ok) ;; P) \lor P)
   by (simp add: impl-alt-def seqr-or-distl)
 also from assms have ... = ((((\neg \$ok) ;; true_h) ;; P) \lor P)
   by (simp add: precond-right-unit unrest)
 also have ... = (((\neg \$ok) ;; (true_h ;; P)) \lor P)
   by (simp add: seqr-assoc)
 also from assms have ... = (\$ok \Rightarrow P)
   by (simp add: H1-left-zero impl-alt-def precond-right-unit unrest)
 finally show ?thesis using assms
   by (simp add: H1-def Healthy-def')
qed
theorem H1-algebraic:
  P \text{ is } H1 \longleftrightarrow (true_h ;; P) = true_h \land (II_D ;; P) = P
 using H1-algebraic-intro H1-left-unit H1-left-zero by blast
theorem H1-nok-left-zero:
 fixes P :: '\alpha \text{ hrel-des}
 assumes P is H1
 shows ((\neg \$ok) ;; P) = (\neg \$ok)
proof -
 have ((\neg \$ok) ;; P) = (((\neg \$ok) ;; true_h) ;; P)
   by (simp add: precond-right-unit unrest)
 also have ... = ((\neg \$ok) ;; true_h)
   by (metis H1-left-zero assms seqr-assoc)
 also have ... = (\neg \$ok)
   by (simp add: precond-right-unit unrest)
 finally show ?thesis.
qed
lemma H1-design:
 H1(P \vdash Q) = (P \vdash Q)
 by (rel-auto)
lemma H1-rdesign:
 H1(P \vdash_r Q) = (P \vdash_r Q)
 by (rel-auto)
lemma H1-choice-closed [closure]:
  \llbracket P \text{ is } H1; Q \text{ is } H1 \rrbracket \Longrightarrow P \sqcap Q \text{ is } H1 \rrbracket
 by (simp add: H1-def Healthy-def' disj-upred-def impl-alt-def semilattice-sup-class.sup-left-commute)
lemma H1-inf-closed [closure]:
  \llbracket P \text{ is } H1; Q \text{ is } H1 \rrbracket \Longrightarrow P \sqcup Q \text{ is } H1
 by (rel-blast)
lemma H1-UINF:
 assumes A \neq \{\}
 shows H1(\bigcap i \in A \cdot P(i)) = (\bigcap i \in A \cdot H1(P(i)))
```

```
using assms by (rel-auto)
lemma H1-Sup:
 assumes A \neq \{\} \ \forall \ P \in A. \ P \ is \ H1
 shows (   A) is H1
proof -
 from assms(2) have H1 ' A = A
   by (auto simp add: Healthy-def rev-image-eqI)
 with H1-UINF[of A id, OF assms(1)] show ?thesis
   by (simp add: UINF-as-Sup-image Healthy-def, presburger)
qed
lemma H1-USUP:
 shows H1(| | i \in A \cdot P(i)) = (| | i \in A \cdot H1(P(i)))
 by (rel-auto)
lemma H1-Inf [closure]:
 assumes \forall P \in A. P \text{ is } H1
 proof -
 from assms have H1 ' A = A
   by (auto simp add: Healthy-def rev-image-eqI)
 with H1-USUP[of A id] show ?thesis
   by (simp add: USUP-as-Inf-image Healthy-def, presburger)
qed
2.2
       H2: A specification cannot require non-termination
definition J :: '\alpha \text{ hrel-des } \mathbf{where}
[upred-defs]: J = ((\$ok \Rightarrow \$ok') \land [II]_D)
definition H2 where
[upred-defs]: H2(P) \equiv P ;; J
lemma J-split:
 shows (P :: J) = (P^f \lor (P^t \land \$ok'))
proof -
 have (P :: J) = (P :: ((\$ok \Rightarrow \$ok') \land \lceil II \rceil_D))
   by (simp add: H2-def J-def design-def)
 also have ... = (P ; ((\$ok \Rightarrow \$ok \land \$ok') \land \lceil II \rceil_D))
   by (rel-auto)
 also have ... = ((P : (\neg \$ok \land [II]_D)) \lor (P : (\$ok \land ([II]_D \land \$ok'))))
   by (rel-auto)
 also have ... = (P^f \lor (P^t \land \$ok'))
 proof -
   have (P :: (\neg \$ok \land \lceil II \rceil_D)) = P^f
   proof -
     have (P :: (\neg \$ok \land \lceil II \rceil_D)) = ((P \land \neg \$ok') :: \lceil II \rceil_D)
       by (rel-auto)
     also have ... = (\exists \$ok' \cdot P \land \$ok' =_u false)
       by (rel-auto)
     also have \dots = P^f
       by (metis C1 one-point out-var-uvar unrest-as-exists ok-vwb-lens vwb-lens-mwb)
    finally show ?thesis.
   moreover have (P : (\$ok \land (\lceil II \rceil_D \land \$ok'))) = (P^t \land \$ok')
```

```
proof -
      have (P :; (\$ok \land ([II]_D \land \$ok'))) = (P :; (\$ok \land II))
       by (rel-auto)
      also have ... = (P^t \wedge \$ok')
       by (rel-auto)
      finally show ?thesis.
   qed
   ultimately show ?thesis
     \mathbf{by} \ simp
  qed
 finally show ?thesis.
qed
lemma H2-split:
 shows H2(P) = (P^f \vee (P^t \wedge \$ok'))
 by (simp add: H2-def J-split)
theorem H2-equivalence:
  P \text{ is } H2 \longleftrightarrow {}^{\iota}P^f \Rightarrow P^t {}^{\iota}
proof -
  have P \Leftrightarrow (P ;; J) \leftrightarrow P \Leftrightarrow (P^f \lor (P^t \land \$ok))
   by (simp add: J-split)
  also have ... \longleftrightarrow (P \Leftrightarrow P^f \vee P^t \wedge \$ok')^f \wedge (P \Leftrightarrow P^f \vee P^t \wedge \$ok')^t
   by (simp add: subst-bool-split)
  also have ... = (P^f \Leftrightarrow P^f) \land (P^t \Leftrightarrow P^f \lor P^t)
   by subst-tac
  also have ... = P^t \Leftrightarrow (P^f \vee P^t)
   by (pred-auto robust)
  also have ... = (P^f \Rightarrow P^t)
   by (pred-auto)
 finally show ?thesis
   by (metis H2-def Healthy-def' taut-iff-eq)
qed
lemma H2-equiv:
  P \ is \ H2 \longleftrightarrow P^t \sqsubseteq P^f
 using H2-equivalence refBy-order by blast
lemma H2-design:
 assumes \$ok' \sharp P \$ok' \sharp Q
 \mathbf{shows}\ \mathit{H2}(P \vdash Q) = P \vdash Q
  using assms
 by (simp add: H2-split design-def usubst unrest, pred-auto)
lemma H2-rdesign:
  H2(P \vdash_r Q) = P \vdash_r Q
 by (simp add: H2-design unrest rdesign-def)
theorem J-idem:
 (J ;; J) = J
 by (rel-auto)
theorem H2-idem:
  H2(H2(P)) = H2(P)
 by (metis H2-def J-idem seqr-assoc)
```

```
theorem H2-Continuous: Continuous H2
 by (rel-auto)
theorem H2-not-okay: H2 (\neg \$ok) = (\neg \$ok)
proof -
 have H2 (\neg \$ok) = ((\neg \$ok)^f \lor ((\neg \$ok)^t \land \$ok'))
   by (simp add: H2-split)
 also have ... = (\neg \$ok \lor (\neg \$ok) \land \$ok')
   by (subst-tac)
 also have ... = (\neg \$ok)
   by (pred-auto)
 finally show ?thesis.
lemma H2-true: H2(true) = true
 by (rel-auto)
lemma H2-choice-closed [closure]:
 \llbracket P \text{ is } H2; Q \text{ is } H2 \rrbracket \Longrightarrow P \sqcap Q \text{ is } H2
 by (metis H2-def Healthy-def' disj-upred-def seqr-or-distl)
lemma H2-inf-closed [closure]:
 assumes P is H2 Q is H2
 shows P \sqcup Q is H2
proof -
 have P \sqcup Q = (P^f \vee P^t \wedge \$ok') \sqcup (Q^f \vee Q^t \wedge \$ok')
   by (metis H2-def Healthy-def J-split assms(1) assms(2))
 moreover have H2(...) = ...
   by (simp add: H2-split usubst, pred-auto)
 ultimately show ?thesis
   by (simp add: Healthy-def)
qed
lemma H2\text{-}USUP:
 shows H2(\bigcap i \in A \cdot P(i)) = (\bigcap i \in A \cdot H2(P(i)))
 by (rel-auto)
theorem H1-H2-commute:
 H1 (H2 P) = H2 (H1 P)
proof -
 have H2 (H1 P) = ((\$ok \Rightarrow P) ;; J)
   by (simp add: H1-def H2-def)
 also have ... = ((\neg \$ok \lor P) ;; J)
   by (rel-auto)
 also have ... = (((\neg \$ok) ;; J) \lor (P ;; J))
   using seqr-or-distl by blast
 also have ... = ((H2 (\neg \$ok)) \lor H2(P))
   by (simp add: H2-def)
 also have ... = ((\neg \$ok) \lor H2(P))
   by (simp add: H2-not-okay)
 also have ... = H1(H2(P))
   by (rel-auto)
 finally show ?thesis by simp
qed
```

## 2.3 Designs as H1-H2 predicates

```
abbreviation H1-H2 :: ('\alpha, '\beta) rel-des \Rightarrow ('\alpha, '\beta) rel-des (H) where
H1-H2 P \equiv H1 (H2 P)
lemma H1-H2-comp: \mathbf{H} = H1 \circ H2
 by (auto)
theorem H1-H2-eq-design:
 \mathbf{H}(P) = (\neg P^f) \vdash P^t
proof -
  have \mathbf{H}(P) = (\$ok \Rightarrow H2(P))
   by (simp add: H1-def)
  also have ... = (\$ok \Rightarrow (P^f \lor (P^t \land \$ok')))
   by (metis H2-split)
  also have ... = (\$ok \land (\neg P^f) \Rightarrow \$ok' \land \$ok \land P^t)
   by (rel-auto)
 also have \dots = (\neg P^f) \vdash P^t
   by (rel-auto)
 finally show ?thesis.
qed
theorem H1-H2-is-design:
 assumes P is H1 P is H2
 shows P = (\neg P^f) \vdash P^t
 using assms by (metis H1-H2-eq-design Healthy-def)
theorem H1-H2-eq-rdesign:
 \mathbf{H}(P) = pre_D(P) \vdash_r post_D(P)
proof -
 have \mathbf{H}(P) = (\$ok \Rightarrow H2(P))
   by (simp add: H1-def Healthy-def')
  also have ... = (\$ok \Rightarrow (P^f \lor (P^t \land \$ok')))
   by (metis H2-split)
  also have ... = (\$ok \land (\neg P^f) \Rightarrow \$ok' \land P^t)
   by (pred-auto)
  also have ... = (\$ok \land (\neg P^f) \Rightarrow \$ok' \land \$ok \land P^t)
   by (pred-auto)
  also have ... = (\$ok \land \lceil pre_D(P) \rceil_D \Rightarrow \$ok' \land \$ok \land \lceil post_D(P) \rceil_D)
   by (simp add: ok-post ok-pre)
  also have ... = (\$ok \land \lceil pre_D(P) \rceil_D \Rightarrow \$ok' \land \lceil post_D(P) \rceil_D)
   by (pred-auto)
  also have ... = pre_D(P) \vdash_r post_D(P)
   by (simp add: rdesign-def design-def)
 finally show ?thesis.
qed
theorem H1-H2-is-rdesign:
 assumes P is H1 P is H2
 shows P = pre_D(P) \vdash_r post_D(P)
 by (metis\ H1-H2-eq-rdesign\ Healthy-def\ assms(1)\ assms(2))
lemma H1-H2-refinement:
  assumes P is H Q is H
 shows P \sqsubseteq Q \longleftrightarrow (`pre_D(P) \Rightarrow pre_D(Q)` \land `pre_D(P) \land post_D(Q) \Rightarrow post_D(P)`)
  by (metis H1-H2-eq-rdesign Healthy-if assms rdesign-refinement)
```

```
lemma H1-H2-refines:
 assumes P is \mathbf{H} Q is \mathbf{H} P \sqsubseteq Q
 shows pre_D(Q) \sqsubseteq pre_D(P) \ post_D(P) \sqsubseteq (pre_D(P) \land post_D(Q))
 using H1-H2-refinement assms refBy-order by auto
lemma H1-H2-idempotent: \mathbf{H} (\mathbf{H} P) = \mathbf{H} P
 by (simp add: H1-H2-commute H1-idem H2-idem)
lemma H1-H2-Idempotent [closure]: Idempotent H
 by (simp add: Idempotent-def H1-H2-idempotent)
lemma H1-H2-monotonic [closure]: Monotonic H
 by (simp add: H1-monotone H2-def mono-def seqr-mono)
lemma H1-H2-Continuous [closure]: Continuous H
 by (simp add: Continuous-comp H1-Continuous H1-H2-comp H2-Continuous)
lemma design-is-H1-H2 [closure]:
 \llbracket \$ok' \sharp P; \$ok' \sharp Q \rrbracket \Longrightarrow (P \vdash Q) \text{ is } \mathbf{H}
 by (simp add: H1-design H2-design Healthy-def')
lemma rdesign-is-H1-H2 [closure]:
 (P \vdash_r Q) is \mathbf{H}
 by (simp add: Healthy-def H1-rdesign H2-rdesign)
lemma top-d-is-H1-H2 [closure]: \top_D is H
 by (simp add: H1-def H2-not-okay Healthy-intro impl-alt-def)
lemma bot-d-is-H1-H2 [closure]: \perp_D is H
 by (simp add: bot-d-def closure unrest)
lemma seq-r-H1-H2-closed [closure]:
 assumes P is H Q is H
 shows (P ;; Q) is H
proof -
 obtain P_1 P_2 where P = P_1 \vdash_r P_2
   by (metis H1-H2-commute H1-H2-is-rdesign H2-idem Healthy-def assms(1))
 moreover obtain Q_1 Q_2 where Q = Q_1 \vdash_r Q_2
  \mathbf{by}\ (\textit{metis H1-H2-commute H1-H2-is-rdesign H2-idem Healthy-def assms}(2))
 moreover have ((P_1 \vdash_r P_2) ;; (Q_1 \vdash_r Q_2)) is H
   by (simp add: rdesign-composition rdesign-is-H1-H2)
 ultimately show ?thesis by simp
qed
lemma UINF-H1-H2-closed [closure]:
 assumes A \neq \{\} \ \forall \ P \in A. \ P \ is \mathbf{H}
 shows (  A) is H1-H2
proof -
 from assms have A: A = H1-H2 ' A
   by (auto simp add: Healthy-def rev-image-eqI)
 by (simp add: UINF-as-Sup-collect)
 also have ... = (  P \in A \cdot (\neg P^f) \vdash P^t )
   by (meson H1-H2-eq-design)
```

```
also have ... = (   P \in A \cdot \neg P^f ) \vdash (  P \in A \cdot P^t )
       by (simp add: design-UINF-mem assms)
    also have ... is H1-H2
       by (simp add: design-is-H1-H2 unrest)
   finally show ?thesis.
qed
definition design-inf :: ('\alpha, '\beta) rel-des set \Rightarrow ('\alpha, '\beta) rel-des (\square_{D^-} [900] 900) where
\bigcap_D A = (if (A = \{\}) then \top_D else \bigcap_A)
abbreviation design-sup :: (\alpha, \beta) rel-des set \Rightarrow (\alpha, \beta) rel-des (\beta rel-des (
\bigsqcup_D A \equiv \bigsqcup A
lemma design-inf-H1-H2-closed:
   assumes \forall P \in A. P \text{ is } \mathbf{H}
   shows (\prod_D A) is H
   apply (auto simp add: design-inf-def closure)
   apply (simp add: H1-def H2-not-okay Healthy-def impl-alt-def)
   apply (metis H1-def Healthy-def UINF-H1-H2-closed assms empty-iff impl-alt-def)
done
lemma design-sup-empty [simp]: \prod_{D} \{\} = \top_{D}
   by (simp add: design-inf-def)
lemma design-sup-non-empty [simp]: A \neq \{\} \Longrightarrow \prod_{D} A = \prod_{D} A
    by (simp add: design-inf-def)
\mathbf{lemma}\ \mathit{USUP-mem-H1-H2-closed}\colon
    assumes \bigwedge i. i \in A \Longrightarrow P i is H
    proof -
    from assms have (\bigsqcup i \in A \cdot P \ i) = (\bigsqcup i \in A \cdot \mathbf{H}(P \ i))
       by (auto intro: USUP-cong simp add: Healthy-def)
    by (meson H1-H2-eq-design)
    also have ... = (\bigcap i \in A \cdot \neg (P i)^f) \vdash (| | i \in A \cdot \neg (P i)^f \Rightarrow (P i)^t)
       by (simp add: design-USUP-mem)
   also have ... is H
       by (simp add: design-is-H1-H2 unrest)
    finally show ?thesis.
qed
\mathbf{lemma}\ \mathit{USUP-ind-H1-H2-closed}\colon
    assumes \bigwedge i. P i is H
    shows (   i \cdot P i ) is H
    using assms USUP-mem-H1-H2-closed[of UNIV P] by simp
lemma Inf-H1-H2-closed:
   assumes \forall P \in A. P \text{ is } \mathbf{H}
   proof -
    from assms have A: A = \mathbf{H} \cdot A
       by (auto simp add: Healthy-def rev-image-eqI)
    by (simp add: USUP-as-Inf-collect)
```

```
also have ... = (   P \in A \cdot (  P^f ) \vdash P^t )
   by (meson H1-H2-eq-design)
 also have ... = (   P \in A \cdot \neg P^f ) \vdash (  P \in A \cdot \neg P^f \Rightarrow P^t )
   by (simp add: design-USUP-mem)
 also have ... is H
   by (simp add: design-is-H1-H2 unrest)
 finally show ?thesis.
qed
lemma rdesign-ref-monos:
 assumes P is \mathbf{H} Q is \mathbf{H} P \sqsubseteq Q
 shows pre_D(Q) \sqsubseteq pre_D(P) \ post_D(P) \sqsubseteq (pre_D(P) \land post_D(Q))
proof -
 have r: P \sqsubseteq Q \longleftrightarrow (`pre_D(P) \Rightarrow pre_D(Q)` \land `pre_D(P) \land post_D(Q) \Rightarrow post_D(P)`)
   by (metis\ H1-H2-eq-rdesign\ Healthy-if\ assms(1)\ assms(2)\ rdesign-refinement)
 from r assms show pre_D(Q) \sqsubseteq pre_D(P)
   by (auto simp add: refBy-order)
 from r assms show post_D(P) \sqsubseteq (pre_D(P) \land post_D(Q))
   by (auto simp add: refBy-order)
\mathbf{qed}
2.4
       H3: The design assumption is a precondition
definition H3::('\alpha, '\beta) \ rel\ des \Rightarrow ('\alpha, '\beta) \ rel\ des \ where
[upred-defs]: H3 (P) \equiv P ;; II_D
theorem H3-idem:
 H3(H3(P)) = H3(P)
 by (metis H3-def design-skip-idem seqr-assoc)
theorem H3-mono:
 P \sqsubseteq Q \Longrightarrow H3(P) \sqsubseteq H3(Q)
 by (simp add: H3-def seqr-mono)
theorem H3-Monotonic:
 Monotonic H3
 by (simp add: H3-mono mono-def)
theorem H3-Continuous: Continuous H3
 by (rel-auto)
theorem design-condition-is-H3:
 assumes out\alpha \sharp p
 shows (p \vdash Q) is H3
proof -
 have ((p \vdash Q) ;; II_D) = (\neg ((\neg p) ;; true)) \vdash (Q^t ;; II[[true/\$ok]])
   by (simp add: skip-d-alt-def design-composition-subst unrest assms)
 also have ... = p \vdash (Q^t ;; II[true/\$ok])
   using assms precond-equiv seqr-true-lemma by force
 also have ... = p \vdash Q
   by (rel-auto)
 finally show ?thesis
   by (simp add: H3-def Healthy-def')
qed
theorem rdesign-H3-iff-pre:
```

```
P \vdash_r Q \text{ is } H3 \longleftrightarrow P = (P ;; true)
proof -
  have (P \vdash_r Q) ;; II_D = (P \vdash_r Q) ;; (true \vdash_r II)
    by (simp add: skip-d-def)
  also have ... = (\neg ((\neg P) ;; true) \land \neg (Q ;; (\neg true))) \vdash_r (Q ;; II)
    by (simp add: rdesign-composition)
  also have ... = (\neg ((\neg P) ;; true) \land \neg (Q ;; (\neg true))) \vdash_r Q
    by simp
  also have ... = (\neg ((\neg P) ;; true)) \vdash_r Q
    by (pred-auto)
  finally have P \vdash_r Q \text{ is } H3 \longleftrightarrow P \vdash_r Q = (\neg ((\neg P) ;; true)) \vdash_r Q
   by (metis H3-def Healthy-def')
  also have ... \longleftrightarrow P = (\neg ((\neg P) ;; true))
    by (metis rdesign-pre)
      thm segr-true-lemma
  also have ... \longleftrightarrow P = (P ;; true)
    by (simp add: seqr-true-lemma)
 finally show ?thesis.
qed
theorem design-H3-iff-pre:
  assumes \$ok \ \sharp \ P \ \$ok \ \sharp \ P \ \$ok \ \sharp \ Q \ \$ok \ \sharp \ Q
  shows P \vdash Q \text{ is } H3 \longleftrightarrow P = (P ;; true)
proof -
 have P \vdash Q = |P|_D \vdash_r |Q|_D
    by (simp add: assms lift-desr-inv rdesign-def)
  moreover hence |P|_D \vdash_r |Q|_D is H3 \longleftrightarrow |P|_D = (|P|_D ;; true)
   using rdesign-H3-iff-pre by blast
  ultimately show ?thesis
    by (metis assms(1,2) drop-desr-inv lift-desr-inv lift-dist-seq aext-true)
qed
theorem H1-H3-commute:
 H1 (H3 P) = H3 (H1 P)
 by (rel-auto)
lemma skip-d-absorb-J-1:
  (II_D ;; J) = II_D
 by (metis H2-def H2-rdesign skip-d-def)
lemma skip-d-absorb-J-2:
  (J ;; II_D) = II_D
proof -
  have (J :: II_D) = ((\$ok \Rightarrow \$ok') \land [II]_D) :: (true \vdash II)
    by (simp add: J-def skip-d-alt-def)
  also have ... = (\exists ok_0 \cdot ((\$ok \Rightarrow \$ok') \land [II]_D)[ (\circ k_0)/(\$ok')] ;; (true \vdash II)[ (\circ k_0)/(\$ok)] )
    by (subst seqr-middle[of ok], simp-all)
  also have ... = ((((\$ok \Rightarrow \$ok') \land [II]_D)[false/\$ok']]; (true \vdash II)[false/\$ok])
                  \vee (((\$ok \Rightarrow \$ok') \land \lceil II \rceil_D) \llbracket true / \$ok' \rrbracket ;; (true \vdash II) \llbracket true / \$ok \rrbracket))
    by (simp add: disj-comm false-alt-def true-alt-def)
  also have ... = ((\neg \$ok \land \lceil II \rceil_D ;; true) \lor (\lceil II \rceil_D ;; \$ok \land \lceil II \rceil_D))
    by (rel-auto)
  also have ... = II_D
    by (rel-auto)
  finally show ?thesis.
```

```
\mathbf{qed}
```

```
lemma H2-H3-absorb:
 H2 (H3 P) = H3 P
 by (metis H2-def H3-def segr-assoc skip-d-absorb-J-1)
lemma H3-H2-absorb:
 H3 (H2 P) = H3 P
 by (metis H2-def H3-def seqr-assoc skip-d-absorb-J-2)
theorem H2-H3-commute:
 H2 (H3 P) = H3 (H2 P)
 by (simp add: H2-H3-absorb H3-H2-absorb)
theorem H3-design-pre:
 assumes \$ok \sharp p \ out \alpha \sharp p \ \$ok \sharp Q \ \$ok ' \sharp Q
 shows H3(p \vdash Q) = p \vdash Q
 using assms
 by (metis Healthy-def' design-H3-iff-pre precond-right-unit unrest-out \alpha-var ok-vwb-lens vwb-lens-mwb)
theorem H3-rdesign-pre:
 assumes out\alpha \sharp p
 shows H3(p \vdash_r Q) = p \vdash_r Q
 using assms
 by (simp\ add:\ H3\text{-}def)
theorem H3-ndesign: H3(p \vdash_n Q) = (p \vdash_n Q)
 by (simp add: H3-def ndesign-def unrest-pre-out\alpha)
theorem ndesign-is-H3 [closure]: p \vdash_n Q is H3
 by (simp add: H3-ndesign Healthy-def)
2.5
       Normal Designs as H1-H3 predicates
A normal design [3] refers only to initial state variables in the precondition.
abbreviation H1-H3 :: ('\alpha, '\beta) rel-des \Rightarrow ('\alpha, '\beta) rel-des (N) where
H1-H3 p \equiv H1 (H3 p)
lemma H1-H3-comp: H1-H3 = H1 \circ H3
 by (auto)
theorem H1-H3-is-design:
 assumes P is H1 P is H3
 shows P = (\neg P^f) \vdash P^t
 by (metis\ H1-H2-eq-design\ H2-H3-absorb\ Healthy-def'\ assms(1)\ assms(2))
theorem H1-H3-is-rdesign:
 assumes P is H1 P is H3
 shows P = pre_D(P) \vdash_r post_D(P)
 by (metis H1-H2-is-rdesign H2-H3-absorb Healthy-def' assms)
theorem H1-H3-is-normal-design:
 assumes P is H1 P is H3
 shows P = \lfloor pre_D(P) \rfloor < \vdash_n post_D(P)
 by (metis H1-H3-is-rdesign assms drop-pre-inv ndesign-def precond-equiv rdesign-H3-iff-pre)
```

```
lemma H1-H3-idempotent: \mathbf{N} (\mathbf{N} P) = \mathbf{N} P
 by (simp add: H1-H3-commute H1-idem H3-idem)
lemma H1-H3-Idempotent [closure]: Idempotent N
 by (simp add: Idempotent-def H1-H3-idempotent)
lemma H1-H3-monotonic [closure]: Monotonic N
 by (simp add: H1-monotone H3-mono mono-def)
lemma H1-H3-Continuous [closure]: Continuous N
 by (simp add: Continuous-comp H1-Continuous H1-H3-comp H3-Continuous)
lemma H1-H3-intro:
 assumes P is \mathbf{H} out\alpha \sharp pre_D(P)
 shows P is N
 by (metis H1-H2-eq-rdesign H1-rdesign H3-rdesign-pre Healthy-def' assms)
lemma H1-H3-impl-H2 [closure]: P is \mathbf{N} \Longrightarrow P is \mathbf{H}
 by (metis H1-H2-commute H1-idem H2-H3-absorb Healthy-def')
lemma H1-H3-eq-design-d-comp: \mathbf{N}(P) = ((\neg P^f) \vdash P^t) :: II_D
 by (metis H1-H2-eq-design H1-H3-commute H3-H2-absorb H3-def)
lemma H1-H3-eq-design: \mathbf{N}(P) = (\neg (P^f :: true)) \vdash P^t
 apply (simp add: H1-H3-eq-design-d-comp skip-d-alt-def)
 apply (subst design-composition-subst)
 apply (simp-all add: usubst unrest)
 apply (rel-auto)
done
lemma H3-unrest-out-alpha-nok [unrest]:
 assumes P is N
 shows out\alpha \ \sharp \ P^f
proof -
 have P = (\neg (P^f ;; true)) \vdash P^t
   by (metis H1-H3-eq-design Healthy-def assms)
 also have out\alpha \sharp (...^f)
   by (simp add: design-def usubst unrest, rel-auto)
 finally show ?thesis.
qed
lemma H3-unrest-out-alpha [unrest]: P is \mathbb{N} \Longrightarrow out\alpha \sharp pre_D(P)
 by (metis H1-H3-commute H1-H3-is-rdesign H1-idem Healthy-def' precond-equiv rdesign-H3-iff-pre)
lemma ndesign-H1-H3 [closure]: p \vdash_n Q is N
 by (simp add: H1-rdesign H3-def Healthy-def' ndesign-def unrest-pre-out\alpha)
lemma ndesign-form: P is \mathbf{N} \Longrightarrow (|pre_D(P)| < \vdash_n post_D(P)) = P
 by (metis H1-H2-eq-rdesign H1-H3-impl-H2 H3-unrest-out-alpha Healthy-def drop-pre-inv ndesign-def)
lemma des-bot-H1-H3 [closure]: \perp_D is N
 by (metis H1-design H3-def Healthy-def' design-false-pre design-true-left-zero skip-d-alt-def bot-d-def)
lemma des-top-is-H1-H3 [closure]: \top_D is N
```

```
by (metis ndesign-H1-H3 ndesign-miracle)
lemma skip-d-is-H1-H3 [closure]: II_D is N
 by (simp add: ndesign-H1-H3 skip-d-ndes-def)
lemma seq-r-H1-H3-closed [closure]:
 assumes P is N Q is N
 shows (P ;; Q) is N
 by (metis (no-types) H1-H2-eq-design H1-H3-eq-design-d-comp H1-H3-impl-H2 Healthy-def assms(1)
assms(2) seq-r-H1-H2-closed seqr-assoc)
lemma dcond-H1-H2-closed [closure]:
 assumes P is \mathbb{N} Q is \mathbb{N}
 shows (P \triangleleft b \triangleright_D Q) is N
 by (metis assms ndesign-H1-H3 ndesign-dcond ndesign-form)
lemma inf-H1-H2-closed [closure]:
 assumes P is N Q is N
 shows (P \sqcap Q) is N
 by (metis assms ndesign-H1-H3 ndesign-choice ndesign-form)
lemma sup-H1-H2-closed [closure]:
 assumes P is N Q is N
 shows (P \sqcup Q) is N
 by (metis assms ndesign-H1-H3 ndesign-inf ndesign-form)
lemma ndes-segr-miracle:
 assumes P is N
 shows P :: \top_D = |pre_D P| < \vdash_n false
 by (simp add: assms ndesign-form ndesign-miracle)
 also have ... = |pre_D P| < \vdash_n false
   by (simp add: ndesign-composition-wp wp alpha)
 finally show ?thesis.
qed
lemma ndes-segr-abort:
 assumes P is N
 shows P :: \perp_D = (|pre_D P|_{<} \land post_D P wp false) \vdash_n false
 have P :: \perp_D = (\lfloor pre_D(P) \rfloor \leftarrow_n post_D(P)) :: (false \vdash_n false)
   by (simp add: assms bot-d-true ndesign-false-pre ndesign-form)
 also have ... = (|pre_D P| < \land post_D P wp false) \vdash_n false
   by (simp add: ndesign-composition-wp alpha)
 finally show ?thesis.
qed
lemma USUP-ind-H1-H3-closed [closure]:
 \llbracket \land i. \ P \ i \ is \ \mathbf{N} \rrbracket \Longrightarrow (|| i \cdot P \ i) \ is \ \mathbf{N}
 by (rule H1-H3-intro, simp-all add: H1-H3-impl-H2 USUP-ind-H1-H2-closed preD-USUP-ind unrest)
2.6
       H4: Feasibility
definition H_4 :: ('\alpha, '\beta) rel-des \Rightarrow ('\alpha, '\beta) rel-des where
```

[upred-defs]:  $H_4(P) = ((P;;true) \Rightarrow P)$ 

```
theorem H_4-idem:
 H_4(H_4(P)) = H_4(P)
 by (rel-auto)
lemma is-H4-alt-def:
  P \text{ is } H4 \longleftrightarrow (P \text{ };; \text{ } true) = true
 by (rel-blast)
end
2.7
       UTP theory of Designs
theory utp-des-theory
 imports utp-des-healths
begin
2.8
       UTP theories
typedecl DES
\mathbf{typedecl}\ \mathit{NDES}
abbreviation DES \equiv UTHY(DES, '\alpha des)
abbreviation NDES \equiv UTHY(NDES, '\alpha des)
overloading
  des-hcond == utp-hcond :: (DES, '<math>\alpha des) uthy \Rightarrow ('\alpha des \times '\alpha des) health
  des\text{-}unit == utp\text{-}unit :: (DES, '\alpha des) uthy \Rightarrow '\alpha hrel-des (unchecked)
 ndes-hcond == utp-hcond :: (NDES, '\alpha des) uthy \Rightarrow ('\alpha des \times '\alpha des) health
 ndes-unit == utp-unit :: (NDES, '\alpha des) uthy \Rightarrow '\alpha hrel-des (unchecked)
begin
 definition des-hcond :: (DES, '\alpha des) uthy \Rightarrow ('\alpha des \times '\alpha des) health where
 [upred-defs]: des-hcond t = H1-H2
 definition des-unit :: (DES, '\alpha des) uthy \Rightarrow '\alpha hrel-des where
 [upred-defs]: des-unit t = II_D
 definition ndes-hcond :: (NDES, '\alpha des) uthy \Rightarrow ('\alpha des \times '\alpha des) health where
  [upred-defs]: ndes-hcond\ t = H1-H3
 definition ndes-unit :: (NDES, '\alpha des) uthy \Rightarrow '\alpha hrel-des where
  [upred-defs]: ndes-unit t = II_D
end
interpretation des-utp-theory: utp-theory DES
 by (simp add: H1-H2-commute H1-idem H2-idem des-hcond-def utp-theory-def)
interpretation ndes-utp-theory: utp-theory NDES
 by (simp add: H1-H3-commute H1-idem H3-idem ndes-hcond-def utp-theory.intro)
interpretation des-left-unital: utp-theory-left-unital DES
 \mathbf{apply} \ (\mathit{unfold-locales})
```

**apply** (simp-all add: des-hcond-def des-unit-def)

```
using seq-r-H1-H2-closed apply blast
 apply (simp add: rdesign-is-H1-H2 skip-d-def)
 apply (metis H1-idem H1-left-unit Healthy-def')
done
interpretation ndes-unital: utp-theory-unital NDES
  apply (unfold-locales, simp-all add: ndes-hcond-def ndes-unit-def)
  using seq-r-H1-H3-closed apply blast
 apply (metis H1-rdesign H3-def Healthy-def' design-skip-idem skip-d-def)
 apply (metis H1-idem H1-left-unit Healthy-def')
 apply (metis H1-H3-commute H3-def H3-idem Healthy-def')
done
interpretation design-theory-continuous: utp-theory-continuous DES
  rewrites \bigwedge P. P \in carrier (uthy-order DES) \longleftrightarrow P \text{ is } \mathbf{H}
  and carrier (uthy-order DES) \rightarrow carrier (uthy-order DES) \equiv \llbracket \mathbf{H} \rrbracket_H \rightarrow \llbracket \mathbf{H} \rrbracket_H
  and [\mathcal{H}_{DES}]_H \to [\mathcal{H}_{DES}]_H \equiv [\mathbf{H}]_H \to [\mathbf{H}]_H
 and le (uthy\text{-}order DES) = op \sqsubseteq
  and eq (uthy\text{-}order\ DES) = op =
  by (unfold-locales, simp-all add: des-hcond-def H1-H2-Continuous utp-order-def)
interpretation normal-design-theory-continuous: utp-theory-continuous NDES
  rewrites \bigwedge P. P \in carrier (uthy-order NDES) \longleftrightarrow P \text{ is } \mathbf{N}
  and carrier (uthy-order NDES) \rightarrow carrier (uthy-order NDES) \equiv [\![ \mathbf{N} ]\!]_H \rightarrow [\![ \mathbf{N} ]\!]_H
  and [\![\mathcal{H}_{NDES}]\!]_H \to [\![\mathcal{H}_{NDES}]\!]_H \equiv [\![\mathbf{N}]\!]_H \to [\![\mathbf{N}]\!]_H
  and le (uthy\text{-}order\ NDES) = op \sqsubseteq
  and A \subseteq carrier (uthy\text{-}order NDES) \longleftrightarrow A \subseteq [\![\mathbf{N}]\!]_H
  and eq (uthy\text{-}order\ NDES) = op =
  by (unfold-locales, simp-all add: ndes-hoond-def H1-H3-Continuous utp-order-def)
lemma design-lat-top: \top_{DES} = \mathbf{H}(false)
  by (simp add: design-theory-continuous.healthy-top, simp add: des-hcond-def)
lemma design-lat-bottom: \perp_{DES} = \mathbf{H}(true)
  by (simp add: design-theory-continuous.healthy-bottom, simp add: des-hcond-def)
lemma ndesign-lat-top: \top_{NDES} = \mathbf{N}(false)
  by (metis ndes-hcond-def normal-design-theory-continuous.healthy-top)
lemma ndesign-lat-bottom: \bot_{NDES} = \mathbf{N}(true)
  by (metis ndes-hcond-def normal-design-theory-continuous.healthy-bottom)
2.9
        Galois Connection
```

Example Galois connection between designs and relations. Based on Jim's example in COM-PASS deliverable D23.5.

```
definition [upred-defs]: Des(R) = \mathbf{H}(\lceil R \rceil_D \land \$ok')
definition [upred-defs]: Rel(D) = |D[true, true/\$ok, \$ok']|_D
lemma Des-design: Des(R) = true \vdash_r R
 by (rel-auto)
lemma Rel-design: Rel(P \vdash_r Q) = (P \Rightarrow Q)
 by (rel-auto)
```

```
interpretation Des-Rel-coretract:
  coretract\ DES \leftarrow \langle Des, Rel \rangle \rightarrow REL
    \bigwedge x. \ x \in carrier \ \mathcal{X}_{DES} \leftarrow \langle Des, Rel \rangle \rightarrow REL = (x \ is \ \mathbf{H}) \ \mathbf{and}
    \bigwedge x. \ x \in carrier \ \mathcal{Y}_{DES} \leftarrow \langle Des, Rel \rangle \rightarrow REL = True \ \mathbf{and}
    \pi_{*DES} \leftarrow \langle Des, Rel \rangle \rightarrow REL = Des and
    \pi^*_{DES} \leftarrow \langle Des, Rel \rangle \rightarrow REL = Rel \text{ and }
    le \ \mathcal{X}_{DES} \leftarrow \langle Des, Rel \rangle \rightarrow REL = op \sqsubseteq and
    le \ \mathcal{Y}_{DES} \leftarrow \langle Des, Rel \rangle \rightarrow \ REL = \ op \ \sqsubseteq
proof (unfold-locales, simp-all add: rel-hcond-def des-hcond-def)
  show \bigwedge x. x is id
    by (simp add: Healthy-def)
next
  show Rel \in [\![\mathbf{H}]\!]_H \to [\![id]\!]_H
    by (auto simp add: Rel-def rel-hcond-def Healthy-def)
  show Des \in [id]_H \to [H]_H
    by (auto simp add: Des-def des-hond-def Healthy-def H1-H2-commute H1-idem H2-idem)
  \mathbf{fix} \ R :: 'a \ hrel
  show R \sqsubseteq Rel (Des R)
    by (simp add: Des-design Rel-design)
  fix R :: 'a \ hrel \ and \ D :: 'a \ hrel-des
  assume a: D is H
  then obtain D_1 D_2 where D: D = D_1 \vdash_r D_2
    by (metis H1-H2-commute H1-H2-is-rdesign H1-idem Healthy-def')
  show (Rel\ D \sqsubseteq R) = (D \sqsubseteq Des\ R)
  proof -
    have (D \sqsubseteq Des R) = (D_1 \vdash_r D_2 \sqsubseteq true \vdash_r R)
      by (simp add: D Des-design)
    also have ... = D_1 \wedge R \Rightarrow D_2
      by (simp add: rdesign-refinement)
    also have ... = ((D_1 \Rightarrow D_2) \sqsubseteq R)
      by (rel-auto)
    also have \dots = (Rel \ D \sqsubseteq R)
      by (simp add: D Rel-design)
    finally show ?thesis ..
  qed
qed
From this interpretation we gain many Galois theorems. Some require simplification to remove
superfluous assumptions.
thm Des-Rel-coretract.deflation[simplified]
thm Des-Rel-coretract.inflation
thm Des-Rel-coretract.upper-comp[simplified]
thm Des-Rel-coretract.lower-comp
           Fixed Points
abbreviation design-lfp :: ('\alpha \ hrel-des \Rightarrow '\alpha \ hrel-des) \Rightarrow '\alpha \ hrel-des \ (\mu_D) where
\mu_D F \equiv \mu_{DES} F
abbreviation design-gfp :: ('\alpha hrel-des \Rightarrow '\alpha hrel-des) \Rightarrow '\alpha hrel-des (\nu_D) where
\nu_D F \equiv \nu_{DES} F
```

```
syntax
  -dmu :: pttrn \Rightarrow logic \Rightarrow logic (\mu_D - \cdot - [0, 10] 10)
  -dnu :: pttrn \Rightarrow logic \Rightarrow logic (\nu_D - \cdot - [0, 10] 10)
translations
  \mu_D X \cdot P == \mu_{CONST\ DES} (\lambda X. P)
  \nu_D X \cdot P == \nu_{CONST\ DES} (\lambda X.\ P)
thm design-theory-continuous. GFP-unfold
\mathbf{thm}\ design\text{-}theory\text{-}continuous. LFP\text{-}unfold
Specialise mu-refine-intro to designs.
lemma design-mu-refine-intro:
  assumes \$ok' \sharp C \$ok' \sharp S (C \vdash S) \sqsubseteq F(C \vdash S) `C \Rightarrow (\mu_D F \Leftrightarrow \nu_D F)`
  shows (C \vdash S) \sqsubseteq \mu_D F
proof -
  from assms have (C \vdash S) \sqsubseteq \nu_D F
    {f thm}\ design\mbox{-}theory\mbox{-}continuous.weak.GFP\mbox{-}upperbound
    by (simp add: design-is-H1-H2 design-theory-continuous.weak.GFP-upperbound)
  with assms show ?thesis
    by (rel-auto, metis (no-types, lifting))
lemma rdesign-mu-refine-intro:
  assumes (C \vdash_r S) \sqsubseteq F(C \vdash_r S) ` \lceil C \rceil_D \Rightarrow (\mu_D F \Leftrightarrow \nu_D F)`
  shows (C \vdash_r S) \sqsubseteq \mu_D F
  using assms by (simp add: rdesign-def design-mu-refine-intro unrest)
lemma H1-H2-mu-refine-intro:
  assumes P is \mathbf{H} P \sqsubseteq F(P) '\lceil pre_D(P) \rceil_D \Rightarrow (\mu_D \ F \Leftrightarrow \nu_D \ F)'
  shows P \sqsubseteq \mu_D F
  by (metis H1-H2-eq-rdesign Healthy-if assms rdesign-mu-refine-intro)
Foundational theorem for recursion introduction using a well-founded relation. Contributed by
Dr. Yakoub Nemouchi.
theorem rdesign-mu-wf-refine-intro:
                  WF: wf R
  assumes
                M: Monotonic F
    and
    and
                H \colon F \in \llbracket \mathbf{H} \rrbracket_H \to \llbracket \mathbf{H} \rrbracket_H
    and induct-step:
    \bigwedge st. \ (P \land [e]_{<} =_u \ll st \gg) \vdash_r Q \sqsubseteq F \ ((P \land ([e]_{<}, \ll st \gg)_u \in_u \ll R \gg) \vdash_r Q)
  shows (P \vdash_r Q) \sqsubseteq \mu_D F
proof -
  \mathbf{fix} \ st
  have (P \land \lceil e \rceil_{<} =_{u} \ll st \gg) \vdash_{r} Q \sqsubseteq \mu_{D} F
  using WF proof (induction rule: wf-induct-rule)
    case (less\ st)
    hence \theta: (P \land (\lceil e \rceil_{<}, \ll st \gg)_u \in_u \ll R \gg) \vdash_r Q \sqsubseteq \mu_D F
      by rel-blast
    {f from}\ M\ H\ design\mbox{-}theory\mbox{-}continuous. LFP\mbox{-}lemma3\ mono\mbox{-}Monotone\mbox{-}utp\mbox{-}order
```

from 0.1 have  $2:(P \land (\lceil e \rceil_{\leq}, \ll st \gg)_u \in_u \ll R \gg) \vdash_r Q \sqsubseteq F (\mu_D F)$ 

**have** 1:  $\mu_D$   $F \sqsubseteq F (\mu_D F)$ 

**by** blast

```
by simp
    have 3: F((P \land (\lceil e \rceil_{<}, \ll st \gg)_u \in_u \ll R \gg) \vdash_r Q) \sqsubseteq F(\mu_D F)
      by (simp \ add: \ 0 \ M \ monoD)
    have 4:(P \land \lceil e \rceil < =_u \ll st \gg) \vdash_r Q \sqsubseteq \dots
      by (rule induct-step)
    show ?case
    using order-trans[OF 3 4] HM design-theory-continuous.LFP-lemma2 dual-order.trans mono-Monotone-utp-order
      by blast
  qed
  }
  thus ?thesis
    by (pred\text{-}simp)
theorem ndesign-mu-wf-refine-intro':
  assumes WF: wf R
    and
                M: Monotonic F
                H \colon F \in \llbracket \mathbf{H} \rrbracket_H \to \llbracket \mathbf{H} \rrbracket_H
    and
    and induct-step:
    \bigwedge st. ((p \land e =_u \ll st \gg) \vdash_n Q) \sqsubseteq F ((p \land (e, \ll st \gg)_u \in_u \ll R \gg) \vdash_n Q)
  shows (p \vdash_n Q) \sqsubseteq \mu_D F
  using assms unfolding ndesign-def
  by (rule-tac rdesign-mu-wf-refine-intro[of R F \lceil p \rceil < e], simp-all add: alpha)
theorem ndesign-mu-wf-refine-intro:
  assumes
                WF: wf R
                M: Monotonic F
    and
                H: F \in [\![\mathbf{N}]\!]_H \to [\![\mathbf{N}]\!]_H
    and
    and induct-step:
    \bigwedge st. \ ((p \ \land \ e =_u \ll st \gg) \vdash_n \ Q) \sqsubseteq F \ ((p \ \land \ (e, \ll st \gg)_u \in_u \ll R \gg) \vdash_n \ Q)
  shows (p \vdash_n Q) \sqsubseteq \boldsymbol{\mu}_{NDES} F
proof -
  \mathbf{fix} \ st
  have (p \land e =_u \ll st \gg) \vdash_n Q \sqsubseteq \mu_{NDES} F
  using WF proof (induction rule: wf-induct-rule)
    case (less\ st)
    hence \theta: (p \land (e, \ll st \gg)_u \in_u \ll R \gg) \vdash_n Q \sqsubseteq \mu_{NDES} F
      by rel-blast
    {\bf from}\ M\ H\ design-theory-continuous. LFP-lemma 3\ mono-Monotone-utp-order
    have 1: \mu_{NDES} F \sqsubseteq F (\mu_{NDES} F)
      by (simp add: mono-Monotone-utp-order normal-design-theory-continuous.LFP-lemma3)
    from 0.1 have 2:(p \land (e, \ll st \gg)_u \in_u \ll R \gg) \vdash_n Q \sqsubseteq F (\mu_{NDES} F)
      by simp
    have 3: F((p \land (e, \ll st))_u \in_u \ll R) \vdash_n Q) \sqsubseteq F(\mu_{NDES} F)
      by (simp \ add: \ 0 \ M \ monoD)
    have 4:(p \land e =_u \ll st \gg) \vdash_n Q \sqsubseteq \dots
      by (rule induct-step)
    show ?case
       using order-trans[OF 3 4] H M normal-design-theory-continuous.LFP-lemma2 dual-order.trans
mono-Monotone-utp-order
      \mathbf{by} blast
  \mathbf{qed}
  }
```

```
thus ?thesis
by (pred-simp)
qed
```

end

## 3 Design Proof Tactics

```
theory utp-des-tactics
imports utp-des-theory
begin
```

The tactics split apart a healthy normal design predicate into its pre-postcondition form, using elimination rules, and then attempt to prove refinement conjectures.

named-theorems ND-elim

```
lemma ndes-elim: [P \text{ is } \mathbf{N}; \ Q(|pre_D(P)| < \vdash_n post_D(P))] \implies Q(P)
 by (simp add: ndesign-form)
lemma ndes-ind-elim: \llbracket \bigwedge i. \ P \ i \ is \ \mathbf{N}; \ Q(\lambda \ i. \ \lfloor pre_D(P \ i) \rfloor < \vdash_n \ post_D(P \ i)) \ \rrbracket \Longrightarrow Q(P)
 by (simp add: ndesign-form)
lemma ndes-split [ND-elim]: \llbracket P \text{ is } \mathbf{N} ; \bigwedge \text{ pre post. } Q(\text{pre } \vdash_n \text{post}) \rrbracket \Longrightarrow Q(P)
 by (metis H1-H2-eq-rdesign H1-H3-impl-H2 H3-unrest-out-alpha Healthy-def drop-pre-inv ndesign-def)
Use given closure laws (cls) to expand normal design predicates
method ndes-expand uses cls = (insert \ cls, (erule \ ND-elim)+)
Expand and simplify normal designs
method ndes-simp uses cls =
  ((ndes-expand cls: cls)?, (simp add: ndes-simp closure alpha usubst unrest wp prod.case-eq-if))
Attempt to discharge a refinement between two normal designs
method ndes-refine uses cls =
  (ndes-simp cls: cls; rule-tac ndesign-refine-intro; (insert cls; rel-simp; auto?))
Attempt to discharge an equality between two normal designs
method ndes-eq uses cls =
  (ndes-simp cls: cls; rule-tac antisym; rule-tac ndesign-refine-intro; (insert cls; rel-simp; auto?))
end
```

## 4 Imperative Programming in Designs

```
theory utp-des-prog
imports utp-des-tactics
begin
```

## 4.1 Assignment

```
definition assigns-d :: '\alpha \ usubst \Rightarrow '\alpha \ hrel-des \ (\langle - \rangle_D) where [upred-defs]: assigns-d \ \sigma = (true \vdash_r \ assigns-r \ \sigma)
```

```
syntax
   -assignmentd :: svids \Rightarrow uexprs \Rightarrow logic (infixr := _D 72)
translations
   -assignmentd xs \ vs => CONST \ assigns-d \ (-mk-usubst \ (CONST \ id) \ xs \ vs)
   -assignmentd x \ v \le CONST assigns-d (CONST subst-upd (CONST id) x \ v)
   -assignmentd \ x \ v \le -assignmentd \ (-spvar \ x) \ v
  x,y:=_D u,v <= CONST \ assigns-d \ (CONST \ subst-upd \ (CONST \ subst-upd \ (CONST \ id) \ (CONST \ svar)
(x) (x)
lemma assigns-d-is-H1-H2 [closure]: \langle \sigma \rangle_D is H
   by (simp add: assigns-d-def rdesign-is-H1-H2)
lemma assigns-d-H1-H3 [closure]: \langle \sigma \rangle_D is N
   by (metis H1-rdesign H3-ndesign Healthy-def' aext-true assigns-d-def ndesign-def)
Designs are closed under substitutions on state variables only (via lifting)
lemma state-subst-H1-H2-closed [closure]:
   P \text{ is } \mathbf{H} \Longrightarrow [\sigma \oplus_s \Sigma_D]_s \dagger P \text{ is } \mathbf{H}
   by (metis H1-H2-eq-rdesign Healthy-if rdesign-is-H1-H2 state-subst-design)
lemma assigns-d-ndes-def [ndes-simp]:
   \langle \sigma \rangle_D = (true \vdash_n \langle \sigma \rangle_a)
   by (rel-auto)
lemma assigns-d-id [simp]: \langle id \rangle_D = II_D
   by (rel-auto)
\mathbf{lemma}\ assign-d-left-comp:
   (\langle f \rangle_D ;; (P \vdash_r Q)) = (\lceil f \rceil_s \dagger P \vdash_r \lceil f \rceil_s \dagger Q)
   by (simp add: assigns-d-def rdesign-composition assigns-r-comp subst-not)
lemma assign-d-right-comp:
   ((P \vdash_r Q) ;; \langle f \rangle_D) = ((\neg ((\neg P) ;; true)) \vdash_r (Q ;; \langle f \rangle_a))
   by (simp add: assigns-d-def rdesign-composition)
lemma assigns-d-comp:
   (\langle f \rangle_D ;; \langle g \rangle_D) = \langle g \circ f \rangle_D
   by (simp add: assigns-d-def rdesign-composition assigns-comp)
lemma assigns-d-comp-ext:
   fixes P :: '\alpha \ hrel-des
   assumes P is H
   shows (\langle \sigma \rangle_D ;; P) = [\sigma \oplus_s \Sigma_D]_s \dagger P
proof -
   have \langle \sigma \rangle_D :: P = \langle \sigma \rangle_D :: (pre_D(P) \vdash_r post_D(P))
      by (metis H1-H2-commute H1-H2-is-rdesign H2-idem Healthy-def' assms)
   also have ... = \lceil \sigma \rceil_s \dagger pre_D(P) \vdash_r \lceil \sigma \rceil_s \dagger post_D(P)
      by (simp add: assign-d-left-comp)
   also have ... = [\sigma \oplus_s \Sigma_D]_s \dagger (pre_D(P) \vdash_r post_D(P))
      by (rel-auto)
   also have ... = [\sigma \oplus_s \Sigma_D]_s \dagger P
      by (metis H1-H2-commute H1-H2-is-rdesign H2-idem Healthy-def' assms)
```

finally show ?thesis.

```
Normal designs are closed under substitutions on state variables only
lemma state-subst-H1-H3-closed [closure]:
  P \text{ is } \mathbf{N} \Longrightarrow [\sigma \oplus_s \Sigma_D]_s \dagger P \text{ is } \mathbf{N}
 by (metis H1-H2-eq-rdesign H1-H3-impl-H2 Healthy-if assign-d-left-comp assigns-d-H1-H3 seq-r-H1-H3-closed
state-subst-design)
lemma H_4-assigns-d: \langle \sigma \rangle_D is H_4
proof -
 have (\langle \sigma \rangle_D : (false \vdash_r true_h)) = (false \vdash_r true)
    by (simp add: assigns-d-def rdesign-composition assigns-r-feasible)
  moreover have \dots = true
    by (rel-auto)
  ultimately show ?thesis
    using is-H4-alt-def by auto
qed
        Guarded Commands
4.2
definition GrdCommD :: '\alpha \ upred \Rightarrow ('\alpha, '\beta) \ rel-des \Rightarrow ('\alpha, '\beta) \ rel-des \ (- \rightarrow_D - [85, 86] \ 85) where
[upred-defs]: b \to_D P = P \triangleleft b \triangleright_D \top_D
lemma GrdCommD-ndes-simp [ndes-simp]:
  b \rightarrow_D (p_1 \vdash_n P_2) = ((b \Rightarrow p_1) \vdash_n (\lceil b \rceil < \land P_2))
```

```
by (rel-auto)
lemma GrdCommD-H1-H3-closed [closure]: P is \mathbb{N} \Longrightarrow b \to_D P is \mathbb{N}
```

```
lemma GrdCommD-true [simp]: true \rightarrow_D P = P
 by (rel-auto)
```

**by** (simp add: GrdCommD-def closure)

```
lemma GrdCommD-false [simp]: false \rightarrow_D P = \top_D
 by (rel-auto)
```

```
lemma GrdCommD-abort [simp]: b \rightarrow_D true = ((\neg b) \vdash_n false)
```

#### Alternation 4.3

```
consts
```

```
:: 'a \ set \Rightarrow ('a \Rightarrow 'p) \Rightarrow ('a \Rightarrow 'r) \Rightarrow 'r \Rightarrow 'r
ualtern-list :: ('a × 'r) list \Rightarrow 'r \Rightarrow 'r
```

```
definition AlternateD :: 'a set \Rightarrow ('a \Rightarrow '\alpha upred) \Rightarrow ('a \Rightarrow ('\alpha, '\beta) rel-des) \Rightarrow ('\alpha, '\beta) rel-des \Rightarrow ('\alpha, '\beta)
'\beta) rel-des where
[upred-defs, ndes-simp]:
AlternateD A g P Q = (\bigcap i \in A \cdot g(i) \rightarrow_D P(i)) \cap (\bigwedge i \in A \cdot \neg g(i)) \rightarrow_D Q
```

This lemma shows that our generalised alternation is the same operator as Marcel Oliveira's definition of alternation when the else branch is abort.

```
\mathbf{lemma}\ AlternateD	ext{-}abort	ext{-}alternate:
  assumes \bigwedge i. P(i) is N
  shows
```

```
AlternateD \ A \ g \ P \perp_D =
  ((\bigvee i \in A \cdot g(i)) \land (\bigwedge i \in A \cdot g(i) \Rightarrow \lfloor pre_D(P i) \rfloor_{<})) \vdash_n (\bigvee i \in A \cdot \lceil g(i) \rceil_{<} \land post_D(P i))
proof (cases\ A = \{\})
  {f case} False
  have AlternateD A g P \perp_D =
        (\bigcap i \in A \cdot g(i) \rightarrow_D (|pre_D(Pi)| < \vdash_n post_D(Pi))) \cap (\bigwedge i \in A \cdot \neg g(i)) \rightarrow_D (false \vdash_n true)
    by (simp add: AlternateD-def ndesign-form bot-d-ndes-def assms)
  also have ... = ((\bigvee i \in A \cdot g(i)) \land (\bigwedge i \in A \cdot g(i) \Rightarrow \lfloor pre_D(P i) \rfloor_{<})) \vdash_n (\bigvee i \in A \cdot \lceil g(i) \rceil_{<} \land post_D(P i))
i))
    by (simp add: ndes-simp False, rel-auto)
  finally show ?thesis by simp
\mathbf{next}
  case True
  thus ?thesis
    by (simp add: AlternateD-def, rel-auto)
definition AlternateD-list :: ('\alpha upred \times ('\alpha, '\beta) rel-des) list \Rightarrow ('\alpha, '\beta) rel-des \Rightarrow ('\alpha, '\beta) rel-des
where
[upred-defs, ndes-simp]:
AlternateD-list xs P =
  AlternateD \{0..< length\ xs\}\ (\lambda\ i.\ map\ fst\ xs\ !\ i)\ (\lambda\ i.\ map\ snd\ xs\ !\ i)\ P
adhoc-overloading
  ualtern AlternateD and
  ualtern-list AlternateD-list
nonterminal gcomm and gcomms
syntax
  -altind-els :: pttrn \Rightarrow logic \Rightarrow logic \Rightarrow logic \Rightarrow logic \Rightarrow logic (if -\in -\cdot -\rightarrow -else - fi)
                 :: pttrn \Rightarrow logic \Rightarrow logic \Rightarrow logic \Rightarrow logic (if - \in - \cdot - \rightarrow - fi)
  -altind
  -qcomm
                    :: logic \Rightarrow logic \Rightarrow gcomm (- \rightarrow - [65, 66] 65)
  -gcomm-nil :: gcomm \Rightarrow gcomms (-)
  -gcomm-cons :: gcomm \Rightarrow gcomms \Rightarrow gcomms (- |/ - [60, 61] 61)
  -qcomm-show :: logic \Rightarrow logic
  -altgcomm-els :: gcomms \Rightarrow logic \Rightarrow logic (if / - /else - /fi)
                   :: gcomms \Rightarrow logic (if / - /fi)
  -altqcomm
translations
  -altind-els x A g P Q => CONST ualtern A (\lambda x. g) (\lambda x. P) Q
  -altind-els x A g P Q \leq CONST ualtern A (\lambda x. g) (\lambda x'. P) Q
  -altind x \ A \ g \ P => CONST \ ualtern \ A \ (\lambda \ x. \ g) \ (\lambda \ x. \ P) \ (CONST \ Orderings.top)
  -altind x \land g \mid P \mid CONST \mid valtern \mid A \mid (\lambda \mid x. \mid g) \mid (\lambda \mid x' \mid P) \mid (CONST \mid Orderings.top)
  -altgcomm \ cs => CONST \ ualtern-list \ cs \ (CONST \ Orderings.top)
  -altgcomm (-gcomm-show \ cs) <= CONST \ ualtern-list \ cs \ (CONST \ Orderings.top)
  -altgcomm-els cs P => CONST ualtern-list cs P
  -altgcomm-els (-gcomm-show cs) P \le CONST ualtern-list cs P
  -gcomm\ g\ P => (g,\ P)
  -gcomm \ g \ P \le -gcomm -show \ (g, \ P)
  -gcomm-cons \ c \ cs => c \ \# \ cs
  -gcomm-cons (-gcomm-show c) (-gcomm-show (d \# cs)) <= -gcomm-show (c \# d \# cs)
  -gcomm-nil\ c => [c]
  -gcomm-nil\ (-gcomm-show\ c) <= -gcomm-show\ [c]
```

```
lemma AlternateD-H1-H3-closed [closure]:
  assumes \bigwedge i. i \in A \Longrightarrow P i \text{ is } \mathbb{N} Q \text{ is } \mathbb{N}
  shows if i \in A \cdot g(i) \to P(i) else Q fi is N
proof (cases\ A = \{\})
  case True
  then show ?thesis
    by (simp add: AlternateD-def closure false-upred-def assms)
next
  case False
  then show ?thesis
    by (simp add: AlternateD-def closure assms)
qed
lemma AltD-ndes-simp [ndes-simp]:
  if i \in A \cdot g(i) \rightarrow (P_1(i) \vdash_n P_2(i)) else Q_1 \vdash_n Q_2 fi
   = ((\bigwedge i \in A \cdot g \ i \Rightarrow P_1 \ i) \land ((\bigwedge i \in A \cdot \neg g \ i) \Rightarrow Q_1)) \vdash_n
     ((\bigvee i \in A \cdot \lceil g \mid i \rceil < \land P_2 \mid i) \lor (\bigwedge i \in A \cdot \neg \lceil g \mid i \rceil <) \land Q_2)
proof (cases\ A = \{\})
  case True
  then show ?thesis by (simp add: AlternateD-def)
next
  case False
  then show ?thesis
    by (simp add: ndes-simp, rel-auto)
ged
declare UINF-upto-expand-first [ndes-simp]
declare UINF-Suc-shift [ndes-simp]
declare USUP-upto-expand-first [ndes-simp]
declare USUP-Suc-shift [ndes-simp]
declare true-upred-def [THEN sym, ndes-simp]
lemma AlternateD-mono-refine:
  assumes \bigwedge i. P i \sqsubseteq Q i R \sqsubseteq S
  shows (if i \in A \cdot g(i) \rightarrow P(i) \text{ else } R \text{ fi}) \sqsubseteq (if i \in A \cdot g(i) \rightarrow Q(i) \text{ else } S \text{ fi})
  using assms by (rel-auto, meson)
lemma Monotonic-AlternateD [closure]:
  \llbracket \bigwedge i. \ Monotonic \ (Fi); \ Monotonic \ G \ \rrbracket \Longrightarrow Monotonic \ (\lambda \ X. \ if \ i \in A \cdot g(i) \rightarrow Fi \ X \ else \ G(X) \ fi)
  by (rel-auto, meson)
lemma AlternateD-eq:
  assumes A = B \land i. i \in A \Longrightarrow g(i) = h(i) \land i. i \in A \Longrightarrow P(i) = Q(i) R = S
  shows if i \in A \cdot g(i) \to P(i) else R fi = if i \in B \cdot h(i) \to Q(i) else S fi
  by (insert assms, rel-blast)
lemma AlternateD-empty:
  if i \in \{\} \cdot g(i) \rightarrow P(i) \text{ else } Q \text{ fi} = Q
  by (rel-auto)
\mathbf{lemma}\ Alternate D\text{-}true\text{-}singleton:
  assumes P is N
  shows if true \rightarrow P fi = P
  by (ndes-eq cls: assms)
```

```
\mathbf{lemma}\ \mathit{AlernateD-no-ind}\colon
 assumes A \neq \{\} P is N Q is N
 shows if i \in A \cdot b \rightarrow P else Q fi = if b \rightarrow P else Q fi
 by (ndes-eq cls: assms)
lemma AlernateD-singleton:
 assumes P k is N Q is N
 shows if i \in \{k\} · b(i) \rightarrow P(i) else Q fi = if b(k) \rightarrow P(k) else Q fi (is ?lhs = ?rhs)
proof -
 have ?lhs = if i \in \{k\} \cdot b(k) \rightarrow P(k) else Q fi
   by (auto intro: AlternateD-eq simp add: assms ndesign-form)
 also have \dots = ?rhs
   by (simp add: AlernateD-no-ind assms closure)
 finally show ?thesis.
qed
lemma AlternateD-commute:
 assumes P is N Q is N
 shows if g_1 \to P \mid g_2 \to Q fi = if g_2 \to Q \mid g_1 \to P fi
 by (ndes-eq cls:assms)
lemma AlternateD-dcond:
 assumes P is N Q is N
 shows if q \to P else Q fi = P \triangleleft q \triangleright_D Q
 by (ndes-eq cls:assms)
lemma AlternateD-cover:
 assumes P is N Q is N
 shows if g \to P else Q fi = if g \to P \mid (\neg g) \to Q fi
 by (ndes-eq cls: assms)
lemma UINF-ndes-expand:
 assumes \bigwedge i. i \in A \Longrightarrow P(i) is N
 shows (\bigcap i \in A \cdot \lfloor pre_D(P(i)) \rfloor \leftarrow post_D(P(i))) = (\bigcap i \in A \cdot P(i))
 by (rule UINF-cong, simp add: assms ndesign-form)
lemma USUP-ndes-expand:
 assumes \bigwedge i. i \in A \Longrightarrow P(i) is N
 by (rule USUP-cong, simp add: assms ndesign-form)
lemma AlternateD-ndes-expand:
 assumes \bigwedge i. i \in A \Longrightarrow P(i) is \mathbb{N} Q is \mathbb{N}
 shows if i \in A \cdot g(i) \rightarrow P(i) else Q fi =
        if i \in A \cdot g(i) \to (|pre_D(P(i))| < \vdash_n post_D(P(i))) else |pre_D(Q)| < \vdash_n post_D(Q) fi
 apply (simp add: AlternateD-def)
 apply (subst UINF-ndes-expand[THEN sym])
  apply (simp add: assms closure)
 apply (ndes-simp cls: assms)
 apply (rel-auto)
 done
lemma AlternateD-ndes-expand':
 assumes \bigwedge i. i \in A \Longrightarrow P(i) is N
```

```
shows if i \in A \cdot g(i) \to P(i) fi = if i \in A \cdot g(i) \to (\lfloor pre_D(P(i)) \rfloor_{<} \vdash_n post_D(P(i))) fi
  apply (simp add: AlternateD-def)
  apply (subst UINF-ndes-expand[THEN sym])
  apply (simp add: assms closure)
  apply (ndes-simp cls: assms)
  apply (rel-auto)
  done
lemma ndesign-ind-form:
  assumes \bigwedge i. P(i) is N
  shows (\lambda \ i. \lfloor pre_D(P(i)) \rfloor < \vdash_n post_D(P(i))) = P
  by (simp add: assms ndesign-form)
lemma AlternateD-insert:
  assumes \bigwedge i. i \in (insert \ x \ A) \Longrightarrow P(i) is \mathbf{N} Q is \mathbf{N}
 shows if i \in (insert \ x \ A) \cdot g(i) \rightarrow P(i) else Q \ fi =
         if g(x) \to P(x)
            (\bigvee i \in A \cdot g(i)) \to if i \in A \cdot g(i) \to P(i) fi
            else Q
         fi (is ?lhs = ?rhs)
proof -
  have ?lhs = if i \in (insert \ x \ A) \cdot g(i) \rightarrow (|pre_D(P(i))| < \vdash_n post_D(P(i))) else (|pre_D(Q)| < \vdash_n post_D(P(i)))
post_D(Q)) fi
    using AlternateD-ndes-expand assms(1) assms(2) by blast
  also
 have ... =
         if g(x) \to (|pre_D(P(x))| < \vdash_n post_D(P(x)))
            (\bigvee i \in A \cdot g(i)) \rightarrow if i \in A \cdot g(i) \rightarrow \lfloor pre_D(P(i)) \rfloor \leftarrow post_D(P(i)) fi
            else \ \lfloor pre_D(Q) \rfloor < \vdash_n post_D(Q)
    by (ndes-simp cls:assms, rel-auto)
  also have \dots = ?rhs
    by (simp add: AlternateD-ndes-expand' ndesign-form assms)
 finally show ?thesis.
qed
        Iteration
4.4
theorem ndesign-iteration-wp [ndes-simp]:
  (p \vdash_n Q) ;; (p \vdash_n Q) \hat{\ } n = ((\bigwedge i \in \{0..n\} \cdot (Q \hat{\ } i) \ wp \ p) \vdash_n Q \hat{\ } Suc \ n)
proof (induct n)
  case \theta
  then show ?case by (rel-auto)
next
  case (Suc\ n) note hyp = this
  have (p \vdash_n Q);; (p \vdash_n Q) \hat{} Suc n = (p \vdash_n Q);; (p \vdash_n Q);; (p \vdash_n Q) \hat{} n
    by (simp add: upred-semiring.power-Suc)
  also have ... = (p \vdash_n Q) ;; ((| \mid i \in \{0..n\} \cdot Q \hat{i} wp p) \vdash_n Q \hat{s}uc n)
    by (simp \ add: \ hyp)
  also have ... = (p \land Q \ wp \ (| \ | \ i \in \{0..n\} \cdot Q \hat{\ } i \ wp \ p)) \vdash_n (Q \ ;; \ Q) \ ;; \ Q \hat{\ } n
    by (simp add: upred-semiring.power-Suc ndesign-composition-wp segr-assoc)
  also have ... = (p \land (| | i \in \{0..n\} \cdot Q \cap Suc \ i \ wp \ p)) \vdash_n (Q ;; Q) ;; Q \cap n
    by (simp add: upred-semiring.power-Suc wp)
  also have ... = (p \land (\bigsqcup i \in \{0..n\}, Q \cap Suc \ i \ wp \ p)) \vdash_n (Q ;; Q) ;; Q \cap n
    by (simp add: USUP-as-Inf-image)
  also have ... = (p \land (\bigsqcup i \in \{1..Suc\ n\}, Q \land i \ wp\ p)) \vdash_n (Q ;; Q) ;; Q \land n
```

```
by (metis (no-types, lifting) One-nat-def image-Suc-atLeastAtMost image-cong image-image)
  also have ... = (Q \hat{\ } 0 \text{ } wp \text{ } p \land (\bigcup i \in \{1..Suc \text{ } n\}. \text{ } Q \hat{\ } i \text{ } wp \text{ } p)) \vdash_n (Q \text{ } ;; \text{ } Q \hat{\ } i \text{ } m) )
    by (simp \ add: wp)
  by (simp add: Iic-Suc-eq-insert-0 atLeast0AtMost conj-upred-def image-Suc-atMost)
  also have ... = ( \bigsqcup i \in \{0..Suc\ n\} \cdot Q \hat{i} \ wp\ p) \vdash_n Q \hat{suc} (Suc\ n)
    by (simp add: upred-semiring.power-Suc USUP-as-Inf-image upred-semiring.mult-assoc)
 finally show ?case.
qed
Overloadable Syntax
consts
                 :: 'a \ set \Rightarrow ('a \Rightarrow 'p) \Rightarrow ('a \Rightarrow 'r) \Rightarrow 'r
  uiterate
  uiterate-list :: ('a \times 'r) list <math>\Rightarrow 'r
syntax
                 :: pttrn \Rightarrow logic \Rightarrow logic \Rightarrow logic \Rightarrow logic (do - \in - \cdot - \rightarrow - od)
  -iterind
  -itergcomm
                   :: gcomms \Rightarrow logic (do - od)
translations
  -iterind x A g P = CONST \text{ uiterate } A (\lambda x. g) (\lambda x. P)
  -iterind x A g P \leq CONST uiterate A (\lambda x. g) (\lambda x'. P)
  -itergcomm \ cs => CONST \ uiterate-list \ cs
  -itergcomm (-gcomm-show \ cs) <= CONST \ uiterate-list \ cs
definition IterateD :: 'a set \Rightarrow ('a \Rightarrow '\alpha upred) \Rightarrow ('a \Rightarrow '\alpha hrel-des) \Rightarrow '\alpha hrel-des where
[upred-defs, ndes-simp]:
IterateD A g P = (\mu_{NDES} X \cdot if i \in A \cdot g(i) \rightarrow P(i) ;; X else II_D fi)
definition IterateD-list :: ('\alpha upred \times '\alpha hrel-des) list \Rightarrow '\alpha hrel-des where
[upred-defs, ndes-simp]:
IterateD-list xs = IterateD \{0... < length xs\} (\lambda i. fst (nth xs i)) (\lambda i. snd (nth xs i))
adhoc-overloading
  uiterate IterateD and
  uiterate-list IterateD-list
lemma IterateD-H1-H3-closed [closure]:
  assumes \bigwedge i. i \in A \Longrightarrow P i is N
 shows do i \in A \cdot g(i) \rightarrow P(i) od is N
proof (cases\ A = \{\})
  case True
  then show ?thesis
    by (simp add: IterateD-def closure assms)
next
  \mathbf{case} \ \mathit{False}
  then show ?thesis
    by (simp add: IterateD-def closure assms)
qed
lemma IterateD-empty:
  do\ i \in \{\} \cdot g(i) \rightarrow P(i)\ od = II_D
 \textbf{by} \ (simp \ add: IterateD-def \ AlternateD-empty \ normal-design-theory-continuous. LFP-const \ skip-d-is-H1-H3)
```

**lemma** *IterateD-list-single-expand*:

```
do b \rightarrow P od = (\boldsymbol{\mu}_{NDES} X \cdot if b \rightarrow P ;; X else II_D fi)
oops
\mathbf{lemma}\ \mathit{IterateD-singleton}:
 assumes P is N
 shows do b \to P od = do i \in \{0\} \cdot b \to P od
 apply (simp add: IterateD-list-def IterateD-def AlernateD-singleton assms)
 apply (subst AlernateD-singleton)
 apply (simp)
 apply (rel-auto)
oops
lemma IterateD-mono-refine:
  assumes
    \bigwedge i. P i is \mathbb{N} \bigwedge i. Q i is \mathbb{N}
    \bigwedge i. P i \sqsubseteq Q i
 shows (do \ i \in A \cdot g(i) \rightarrow P(i) \ od) \sqsubseteq (do \ i \in A \cdot g(i) \rightarrow Q(i) \ od)
 apply (simp add: IterateD-def normal-design-theory-continuous.utp-lfp-def)
 apply (subst normal-design-theory-continuous.utp-lfp-def)
  apply (simp-all add: closure assms)
 apply (subst normal-design-theory-continuous.utp-lfp-def)
 apply (simp-all add: closure assms)
 apply (simp add: ndes-hcond-def)
 apply (rule gfp-mono)
 apply (rule AlternateD-mono-refine)
 apply (simp-all add: closure segr-mono assms)
done
lemma IterateD-single-refine:
 assumes
    P \text{ is } \mathbf{N} \text{ } Q \text{ is } \mathbf{N} \text{ } P \sqsubseteq Q
 shows (do \ g \rightarrow P \ od) \sqsubseteq (do \ g \rightarrow Q \ od)
oops
\mathbf{lemma}\ \mathit{IterateD-refine-intro}:
 fixes V :: (nat, 'a) \ uexpr
 assumes vwb-lens w
 shows
  I \vdash_n (w{:}[\lceil I \land \neg (\bigvee i{\in}A \boldsymbol{\cdot} g(i))\rceil_{>}]) \sqsubseteq
  do\ i \in A \cdot g(i) \to (I \wedge g(i)) \vdash_n (w: [\lceil I \rceil_{>} \wedge \lceil V \rceil_{>} <_u \lceil V \rceil_{<}]) \ od
proof (cases\ A = \{\})
 {f case}\ {\it True}
  with assms show ?thesis
    by (simp add: IterateD-empty, rel-auto)
next
  case False
  then show ?thesis
  using assms
    apply (simp add: IterateD-def)
    apply (rule ndesign-mu-wf-refine-intro[where e=V and R=\{(x, y), x < y\}])
    apply (simp-all add: wf closure)
    apply (simp add: ndes-simp unrest)
    apply (rule ndesign-refine-intro)
    apply (rel-auto)
    apply (rel-auto)
```

```
apply (metis mwb-lens.put-put vwb-lens-mwb)
  done
qed
{\bf lemma}\ IterateD	ext{-}single	ext{-}refine	ext{-}intro:
  fixes V :: (nat, 'a) \ uexpr
  assumes vwb-lens w
  shows
  I \vdash_n (w:[[I \land \neg g]_>]) \sqsubseteq
  do\ g \to ((I \land g) \vdash_n (w:[\lceil I \rceil_{>} \land \lceil V \rceil_{>} <_u \lceil V \rceil_{<}]))\ od
  apply (rule order-trans)
  defer
  apply (rule IterateD-refine-intro[of w \{0\} \lambda i. g I V, simplified, OF assms(1)])
4.5
         Let and Local Variables
definition LetD :: ('a, '\alpha) uexpr \Rightarrow ('a \Rightarrow '\alpha \text{ hrel-des}) \Rightarrow '\alpha \text{ hrel-des} where
[upred-defs]: Let D v P = (P x) [x \rightarrow [v]_{D < x}]
syntax
                 :: [letbinds, 'a] \Rightarrow 'a
  -LetD
                                                             ((let_D (-)/in (-)) [0, 10] 10)
translations
  -LetD (-binds b bs) e \rightleftharpoons -LetD b (-LetD bs e)
                               \Rightarrow CONST LetD a (\lambda x. e)
  let_D x = a in e
lemma LetD-ndes-simp [ndes-simp]:
  Let D\ v\ (\lambda\ x.\ p(x)\vdash_n\ Q(x))=(p(x)[\![x\to v]\!])\vdash_n\ (Q(x)[\![x\to\lceil v\rceil_<]\!])
  by (rel-auto)
lemma LetD-H1-H3-closed [closure]:
  \llbracket \bigwedge x. \ P(x) \ is \ \mathbf{N} \ \rrbracket \Longrightarrow LetD \ v \ P \ is \ \mathbf{N}
  by (rel-auto)
         Deep Local Variables
4.6
\mathbf{definition}\ \mathit{des-local-state}\ ::
  'a::countable itself \Rightarrow ((nat, 's) local-scheme des, 's, nat, 'a::countable) local-prim where
  des-local-state t = (sstate = \Sigma_D, sassigns = assigns-d, inj-local = nat-inj-univ)
syntax
  -des-local-state-type :: type \Rightarrow logic (\mathcal{L}_D[-])
  -des-var-scope-type :: id \Rightarrow type \Rightarrow logic \ (var_D - :: - \cdot - [0, 0, 10] \ 10)
translations
  \mathcal{L}_D['a] == CONST \ des-local-state \ TYPE('a)
  -des-var-scope-type x \ t \ P = > -var-scope-type (-des-local-state-type t) \ x \ t \ P
  var_D x :: 'a \cdot P \le var[\mathcal{L}_D['a]] x \cdot P
lemma get-rel-local [lens-defs]:
  get_{\mathbf{S}_{\mathcal{L}_D}['a::countable]} = get_{\Sigma_D}
  by (simp add: des-local-state-def)
lemma des-local-state [simp]: utp-local-state \mathcal{L}_D['a::countable]
  \mathbf{by}\ (\mathit{unfold-locales},\ \mathit{simp-all}\ \mathit{add}\colon \mathit{upred-defs}\ \mathit{assigns-comp}\ \mathit{des-local-state-def},\ \mathit{rel-auto})
```

```
(metis\ local.cases-scheme)
lemma sassigns-des-state [simp]: \langle \sigma \rangle_{\mathcal{L}_D['a::countable]} = \langle \sigma \rangle_D
  by (simp add: des-local-state-def)
lemma des-var-open-H1-H3-closed [closure]:
  open[\mathcal{L}_D['a::countable]] is N
  by (simp add: utp-local-state.var-open-def closure)
lemma des-var-close-H1-H3-closed [closure]:
  close[\mathcal{L}_D['a::countable]] is N
  by (simp add: utp-local-state.var-close-def closure)
lemma unrest-ok-vtop-des [unrest]: ok \sharp top[\mathcal{L}_D['a::countable]]
  by (simp add: utp-local-state.top-var-def, simp add: des-local-state-def unrest)
lemma msubst-H1-H3-closed [closure]:
  \llbracket \$ok \sharp v; out\alpha \sharp v; (\bigwedge x. P x is \mathbf{N}) \rrbracket \Longrightarrow (P(x)\llbracket x \rightarrow v \rrbracket) is \mathbf{N}
  by (rel-auto, metis+)
lemma var-block-H1-H3-closed [closure]:
  (\bigwedge x. \ P \ x \ is \ \mathbf{N}) \Longrightarrow \mathcal{V}[\mathcal{L}_D['a::countable], \ P] \ is \ \mathbf{N}
  by (simp add: utp-local-state.var-scope-def closure unrest)
lemma inj-local-rel [simp]: inj-local R_l = \mathcal{U}_{\mathbb{N}}
  by (simp add: rel-local-state-def)
lemma sstate-rel [simp]: \mathbf{s}_{R_l} = 1_L
  by (simp add: rel-local-state-def)
lemma inj-local-des [simp]:
  inj-local \mathcal{L}_D['a::countable] = \mathcal{U}_{\mathbb{N}}
  by (simp add: des-local-state-def)
lemma sstate-des [simp]: \mathbf{s}_{\mathcal{L}_D['a::countable]} = \Sigma_D
  by (simp add: des-local-state-def)
lemma ndesign-msubst-top [usubst]:
  (p \ x \vdash_n Q x)[x \to \lceil top[\mathcal{L}_D['a::countable]]]_{<}] = ((p \ x)[x \to top[R_l['a]]] \vdash_n (Q \ x)[x \to \lceil top[R_l['a]]]_{<}])
 by (rel-auto')
First attempt at a law for expanding design variable blocks. Far from adequate at the moment
though.
lemma ndesign-local-expand-1 [ndes-simp]:
  (var_D \ x :: 'a :: countable \cdot p(x) \vdash_n Q(x)) =
       ( \bigsqcup \ v \cdot (p \ x) \llbracket x \rightarrow top[R_l] \rrbracket \llbracket \&store \ \widehat{\ \ }_u \ \langle \lessdot v \gg \rangle / store \rrbracket ) \vdash_n
        \#_u(\&store) \triangleright \&store)
 apply (simp add: utp-local-state.var-scope-def utp-local-state.var-open-def utp-local-state.var-close-def
seq-UINF-distr' usubst)
  apply (simp add: ndes-simp wp unrest)
 apply (rel-auto')
 done
```

 $\mathbf{end}$ 

## 5 Design Weakest Preconditions

```
theory utp-des-wp
 imports utp-des-prog
begin
definition wp\text{-}design: ('\alpha, '\beta) \text{ } rel\text{-}des \Rightarrow '\beta \text{ } cond \Rightarrow '\alpha \text{ } cond \text{ } (\text{infix } wp_D \text{ } 60) \text{ } \text{where}
[upred-defs]: Q wp_D r = (|pre_D(Q)|; true :: ('\alpha, '\beta) urel|_{<} \land (post_D(Q) wp r))
If two normal designs have the same weakest precondition for any given postcondition, then the
two designs are equivalent.
theorem wpd-eq-intro: \llbracket \bigwedge r. \ (p_1 \vdash_n Q_1) \ wp_D \ r = (p_2 \vdash_n Q_2) \ wp_D \ r \ \rrbracket \Longrightarrow (p_1 \vdash_n Q_1) = (p_2 \vdash_n Q_2)
apply (rel-simp robust; metis curry-conv)
done
theorem wpd-H3-eq-intro: P is H1-H3; Q is H1-H3; A r. P wp_D r = Q wp_D r r \Longrightarrow P = Q
 by (metis H1-H3-commute H1-H3-is-normal-design H3-idem Healthy-def' wpd-eq-intro)
lemma wp-assigns-d [wp]: \langle \sigma \rangle_D wp_D r = \sigma \dagger r
 by (rel-auto)
theorem rdesign-wp [wp]:
  (\lceil p \rceil_{<} \vdash_{r} Q) wp_{D} r = (p \land Q wp r)
 by (rel-auto)
theorem ndesign-wp [wp]:
  (p \vdash_n Q) wp_D r = (p \land Q wp r)
 by (simp add: ndesign-def rdesign-wp)
theorem wpd-seq-r:
  fixes Q1 Q2 :: '\alpha hrel
  shows ((\lceil p1 \rceil_{\leq} \vdash_r Q1) ;; (\lceil p2 \rceil_{\leq} \vdash_r Q2)) wp_D r = (\lceil p1 \rceil_{\leq} \vdash_r Q1) wp_D ((\lceil p2 \rceil_{\leq} \vdash_r Q2) wp_D r)
 apply (simp add: wp)
 apply (subst rdesign-composition-wp)
 apply (simp only: wp)
 apply (rel-auto)
done
theorem wpnd\text{-}seq\text{-}r [wp]:
 fixes Q1 Q2 :: '\alpha hrel
 shows ((p1 \vdash_n Q1) ;; (p2 \vdash_n Q2)) wp_D r = (p1 \vdash_n Q1) wp_D ((p2 \vdash_n Q2) wp_D r)
 by (simp add: ndesign-def wpd-seq-r)
theorem wpd-seq-r-H1-H3 [wp]:
 fixes P Q :: '\alpha hrel-des
 assumes P is N Q is N
 shows (P :; Q) wp_D r = P wp_D (Q wp_D r)
 by (metis H1-H3-commute H1-H3-is-normal-design H1-idem Healthy-def' assms(1) assms(2) wpnd-seq-r)
end
```

## 6 Refinement Calculus

```
theory utp-des-refcalc
imports utp-des-prog
```

## begin

```
definition des-spec :: ('a \Longrightarrow '\alpha) \Rightarrow '\alpha \ upred \Rightarrow ('\alpha \Rightarrow '\alpha \ upred) \Rightarrow '\alpha \ hrel-des where
[upred-defs]: des-spec x \ p \ q = (\bigsqcup v \cdot ((p \land \& \mathbf{v} =_u \ll v \gg) \vdash_n x : [\lceil q(v) \rceil >]))
syntax
  -init-var
                  :: logic
                  :: salpha \Rightarrow logic \Rightarrow logic \Rightarrow logic (-:[-,/-]_D [99,0,0] 100)
  -des-log-const :: pttrn \Rightarrow logic \Rightarrow logic (con_D - \cdot - [0, 10] 10)
translations
  -des-spec x p q =  CONST des-spec x p (\lambda -init-var. q)
  -des-spec (-salphaset (-salphamk x)) p \ q \le CONST \ des-spec x \ p \ (\lambda \ iv. \ q)
  -des-log-const x <math>P = > | | x \cdot P|
parse-translation \langle \! \langle
 fun\ init-var-tr\ [] = Syntax.free\ iv
    | init-var-tr - = raise Match;
[(@{syntax-const -init-var}, K init-var-tr)]
end
\rangle\!\rangle
abbreviation choose_D x \equiv \{\&x\}:[true,true]_D
lemma des-spec-simple-def:
  x:[pre,post]_D = (pre \vdash_n x:[\lceil post \rceil_>])
 by (rel-auto)
lemma des-spec-abort:
 x:[false,post]_D = \bot_D
 by (rel-auto)
lemma des-spec-skip: \emptyset:[true,true]<sub>D</sub> = II_D
 by (rel-auto)
lemma des-spec-strengthen-post:
  assumes 'post' \Rightarrow post'
  shows w:[pre, post]_D \sqsubseteq w:[pre, post]_D
  using assms by (rel-auto)
lemma des-spec-weaken-pre:
  assumes 'pre \Rightarrow pre''
  shows w:[pre, post]_D \sqsubseteq w:[pre', post]_D
  using assms by (rel-auto)
lemma des-spec-refine-skip:
 assumes vwb-lens w 'pre \Rightarrow post'
 shows w:[pre, post]_D \sqsubseteq II_D
 using assms by (rel-auto)
lemma rc-iter:
 fixes V :: (nat, 'a) \ uexpr
 assumes vwb-lens w
```

```
shows w:[ivr, ivr \land \neg (\bigvee i \in A \cdot g(i))]_D
\sqsubseteq (do i \in A \cdot g(i) \rightarrow \bigsqcup iv \cdot w:[ivr \land g(i) \land \ll iv \gg =_u \& \mathbf{v}, ivr \land (V <_u V[\![\ll iv \gg /\mathbf{v}]\!])]_D \ od) \ (\mathbf{is})
?lhs \sqsubseteq ?rhs)
apply (rule order-trans)
defer
apply (simp add: des-spec-simple-def)
apply (rule IterateD-refine-intro[of - - - - V])
apply (simp add: assms)
apply (rule IterateD-mono-refine)
apply (rule IterateD-mono-refine)
apply (rel-auto)
using assms
apply (rel-auto)
done
```

## 7 Theory of Invariants

```
theory utp-des-invariants
imports utp-des-theory
begin
```

The theory of invariants formalises operation and state invariants based on the theory of designs. For more information, please see the associated paper [1, Section 4].

## 7.1 Operation Invariants

```
definition OIH(\psi)(D) = (D \wedge (\$ok \wedge \neg D^f \Rightarrow \psi))
declare OIH-def [upred-defs]
lemma OIH-design:
 assumes D is H1-H2
  shows OIH(\psi)(D) = ((\neg D^f) \vdash (D^t \land \psi))
  have OIH(\psi)(D) = (((\neg D^f) \vdash D^t) \land (\$ok \land \neg D^f \Rightarrow \psi))
   by (metis H1-H2-commute H1-H2-is-design H1-idem Healthy-def' OIH-def assms)
  also have ... = ((\$ok \land \neg D^f \Rightarrow \$ok \land D^t) \land (\$ok \land \neg D^f \Rightarrow \psi))
   by (simp add: design-def)
  also have ... = ((\neg D^f) \vdash (D^t \land \psi))
   by (pred-auto)
 finally show ?thesis.
qed
lemma OIH-idem:
  assumes D is H1-H2 \$ok' <math>\sharp \psi
 shows OIH(\psi)(OIH(\psi)(D)) = OIH(\psi)(D)
  by (simp add: OIH-design design-is-H1-H2 unrest) (simp add: design-def usubst, rel-auto)
lemma OIH-of-design:
  \$ok' \sharp P \Longrightarrow OIH(\psi)(P \vdash Q) = (P \vdash (Q \land \psi))
  by (simp add: OIH-def design-def usubst, rel-auto)
```

#### 7.2 State Invariants

```
definition ISH(\psi)(D) = (D \vee (\$ok \wedge \neg D^f \wedge \lceil \psi \rceil \prec \Rightarrow \$ok' \wedge D^t))
declare ISH-def [upred-defs]
lemma ISH-design: ISH(\psi)(D) = (\neg D^f \land [\psi]_{<}) \vdash D^t
  by (rel-auto, metis+)
lemma ISH-idem: ISH(\psi)(ISH(\psi)(D)) = ISH(\psi)(D)
  by (simp add: ISH-design usubst design-def, pred-auto)
lemma ISH-of-design:
  \llbracket \$ok' \sharp P; \$ok' \sharp Q \rrbracket \Longrightarrow ISH(\psi)(P \vdash Q) = ((P \land \llbracket \psi \rrbracket_{<}) \vdash Q)
  by (simp add: ISH-design design-def usubst, pred-auto)
definition OSH(\psi)(D) = (D \wedge (\$ok \wedge \neg D^f \wedge [\psi]_{<} \Rightarrow [\psi]_{>}))
declare OSH-def [upred-defs]
lemma OSH-as-OIH:
  OSH(\psi)(D) = OIH([\psi]_{<} \Rightarrow [\psi]_{>})(D)
  by (simp add: OSH-def OIH-def, pred-auto)
lemma OSH-design:
  assumes D is H1-H2
  shows OSH(\psi)(D) = ((\neg D^f) \vdash (D^t \land (\lceil \psi \rceil_{<} \Rightarrow \lceil \psi \rceil_{>})))
  by (simp add: OSH-as-OIH OIH-design assms)
lemma OSH-of-design:
  \llbracket \$ok' \sharp P; \$ok' \sharp Q \rrbracket \Longrightarrow OSH(\psi)(P \vdash Q) = (P \vdash (Q \land (\lceil \psi \rceil_{<} \Rightarrow \lceil \psi \rceil_{>})))
  by (simp add: OSH-design design-is-H1-H2 unrest, simp add: design-def usubst, pred-auto)
definition SIH(\psi) = ISH(\psi) \circ OSH(\psi)
declare SIH-def [upred-defs]
lemma SIH-of-design:
  \llbracket \$ok' \sharp P; \$ok' \sharp Q; ok \sharp \psi \rrbracket \Longrightarrow SIH(\psi)(P \vdash Q) = ((P \land \lceil \psi \rceil_{<}) \vdash (Q \land \lceil \psi \rceil_{>}))
  \mathbf{by}\ (simp\ add\colon SIH\text{-}def\ OSH\text{-}of\text{-}design\ ISH\text{-}of\text{-}design\ unrest},\ pred\text{-}auto)
end
```

## 8 Meta Theory for UTP Designs

```
theory utp-designs
imports
utp-des-core
utp-des-healths
utp-des-theory
utp-des-tactics
utp-des-prog
utp-des-wp
utp-des-refcalc
utp-des-invariants
```

## begin end

## References

- [1] A. Cavalcanti, A. Wellings, and J. Woodcock. The Safety-Critical Java memory model formalised. Formal Aspects of Computing, 25(1):37–57, 2012.
- [2] A. Cavalcanti and J. Woodcock. A tutorial introduction to designs in unifying theories of programming. In *Proc. 4th Intl. Conf. on Integrated Formal Methods (IFM)*, volume 2999 of *LNCS*, pages 40–66. Springer, 2004.
- [3] W. Guttman and B. Möller. Normal design algebra. *Journal of Logic and Algebraic Programming*, 79(2):144–173, February 2010.
- [4] T. Hoare and J. He. Unifying Theories of Programming. Prentice-Hall, 1998.